

## Werte suchen in Excel

### Inhalt

Daten auswählen und auswerten mit der Excel-Funktion GRUPPIERENNACH()	2
Mit SPALTENWAHL() gezielt Spalten in einer Excel-Tabelle auswählen	12
So wählen Sie mit ZEILENWAHL() einzelne Zeilen einer Excel-Tabelle aus	16
Mit SVERWEIS() Einträge in Excel-Tabellen suchen	21
Excel-Funktion SVERWEIS() bei Groß- und Kleinschreibung	26
Mit WVERWEIS() Excel-Tabellen durchsuchen	30
Mit der Excel-Funktion INDEX() Tabellenbereiche durchsuchen	35
Wert in einer Excel-Tabelle suchen und Zelle anzeigen lassen	42
Suchen nach ungefähre Übereinstimmung in Excel-Listen	44
Mit der Excel-Funktion EINDEUTIG() Listen auswerten	48
Mit der Excel-Funktion ERSTERWERT() Listen einfach auswerten	53
Wert aus Zelle übernehmen mit Excel-Funktion ÜBERNEHMEN()	58
Excel-Werte auflisten, die bestimmtes Kriterium erfüllen	62
Zelladressen unter Auswahlbedingung in Excel anzeigen	68
Suchfunktion für die exakte Suche in Excel nutzen	73

# Daten auswählen und auswerten mit der Excel-Funktion GRUPPIERENNACH()

Die Funktion GRUPPIERENNACH() bietet Excel-Anwendern eine einfache Möglichkeit, große Datenmengen zu analysieren, indem sie automatisch Daten auswählt, aggregiert, kategorisiert und zusammenfasst – alles mit nur einer einzigen Formel!

Zuletzt geändert am 18.03.2026



Mit der Funktion GRUPPIERENNACH() werten Sie Daten aus und stellen diese wie mit einer Pivot-Tabelle übersichtlich dar. Mit einer einzigen Funktion

- wählen Sie die gewünschten Daten,
- gruppieren die Daten und
- berechnen Summen, Mittelwerte und andere Ergebnisse.

In diesem Beitrag lernen Sie die Funktionsweise von GRUPPIERENNACH() kennen. Sie sehen dies an einem Praxisbeispiel und Sie erfahren, warum diese Funktion ein Must-have für jeden Excel-Anwender ist.

## Was ist die Funktion GRUPPIERENNACH()?

Die Funktion **GRUPPIERENNACH()** ist eine dynamische **Matrixfunktion**, die speziell dafür entwickelt wurde, Daten auszuwählen, zu gruppieren und zusammenzufassen. Sie funktioniert ähnlich wie eine Pivot-Tabelle, jedoch direkt in einer Zelle – ohne manuelle Schritte oder komplizierten Tabellenaufbau.

Mit dieser Funktion können Sie:

- Daten nach definierten Kriterien auswählen,
- gruppieren nach einer oder mehreren Spalten,
- Werte automatisch summieren, zählen, den Durchschnitt berechnen oder andere Aggregationen durchführen,
- große Datensätze schnell übersichtlich zusammenfassen.

GRUPPIERENNACH() besitzt folgende Syntax:

**=GRUPPIERENNACH(row\_fields; values; function; [field\_headers];  
[total\_depth]; [sort\_order]; [filter\_array]; [field\_relationship])**

**row\_fields:** Ein spaltenorientiertes Array oder ein Bereich, der die Werte enthält, die zum Gruppieren von Zeilen und Generieren von Zeilenüberschriften verwendet werden. Dieser Bereich kann mehrere Spalten enthalten. Wenn ja, enthält die Ausgabe Gruppen und Untergruppen (Zeilengruppenebenen).

**values:** Ein spaltenorientiertes Array oder ein Bereich der zu aggregierenden Daten. Diese Daten (Werte) werden zu den Gruppen zusammengefasst und berechnet (siehe function). Das Array oder der Bereich kann mehrere Spalten enthalten. Wenn ja, enthält die Ausgabe mehrere Aggregationen.

**function:** Eine Funktion wie SUMME, MITTELWERT oder ANZAHL, die zur Aggregation von Werten verwendet wird. Es kann auch ein Vektor von Lambdas bereitgestellt werden. Wenn ja, enthält die Ausgabe mehrere Aggregationen. Die Ausrichtung des Vektors bestimmt, ob sie zeilen- oder spaltenweise angeordnet sind.

**field\_headers (optional):** Eine Zahl, die angibt, ob die Spalte(n) mit den row\_fields Header (Spaltenüberschriften) aufweisen und ob die Überschriften in den Ergebnissen angezeigt werden sollen. Mögliche Werte sind:

- Fehlt: Automatisch (Standard)
- 0: Nein, enthält keine Spaltenüberschrift
- 1: Ja, enthält Spaltenüberschrift, aber nicht anzeigen
- 2: Nein, enthält keine Spaltenüberschrift, aber Spaltenüberschrift generieren
- 3: Ja, enthält Spaltenüberschrift und anzeigen

**total\_depth (optional):** Bestimmt, ob die Zeilenüberschriften Summen enthalten sollen. Die folgenden Werte sind möglich:

- Fehlt: Automatisch, Gesamtsummen und nach Möglichkeit Zwischensummen (Standard)
- 0: Keine Gesamtsummen
- 1: Gesamtsummen
- 2: Gesamtsummen und Zwischensummen

**sort\_order (optional):** Eine Zahl, die angibt, wie Zeilen sortiert werden sollen.

Sortiert werden die Gruppen, also die Werte in row\_fields. Gibt es mehrere Spalten für die Gruppierung (Gruppen und Untergruppen), dann gibt die Zahl bei sort\_order an, welche Spalte für die Sortierung gewählt wird. Die folgenden Werte sind möglich:

- Fehlt: Automatisch, die Sortierung erfolgt aufsteigend oder alphabetisch nach der ersten Spalte (Standard)
- 1: die erste Spalte aus row\_fields wird aufsteigend, alphabetisch von A bis Z, sortiert
- -1: die erste Spalte aus row\_fields wird absteigend, alphabetisch von Z bis A, sortiert
- 2: die zweite Spalte aus row\_fields wird aufsteigend, alphabetisch von A bis Z, sortiert
- -2: die zweite Spalte aus row\_fields wird absteigend, alphabetisch von A bis Z, sortiert
- ...

**filter\_array (optional):** Ein spaltenorientiertes Array boolescher Daten, das angibt, ob die entsprechende Datenzeile berücksichtigt werden soll. Sie können eine Bedingung formulieren, deren Ergebnis wahr oder falsch ist. Zeilen (Datensätze), die als Ergebnis „falsch“ haben, werden bei der Ausgabe nicht angezeigt.

**field\_relationship (optional):** Gibt die Beziehungsfelder an, wenn mehrere Spalten für row\_fields bereitgestellt werden. Mögliche Werte sind:

- 0: Hierarchie (Standard); bei einer Hierarchiefeldbeziehung berücksichtigt die Sortierung späterer Feldspalten die Hierarchie früherer Spalten.
- 1: Tabelle; bei einer Tabellenfeldbeziehung erfolgt die Sortierung jeder Feldspalte unabhängig, ohne Hierarchie. Teilergebnisse werden nicht unterstützt, da sie von den Daten mit einer Hierarchie abhängen.

Was hier zunächst mit der allgemeinen Syntax der GRUPPIERENNACH-Funktion beschrieben ist, lässt sich anhand des folgenden Beispiels besser verstehen.

## Praxisbeispiel: Umsatzanalyse für ein Unternehmen

Angenommen, Sie sind für die Analyse der monatlichen Umsätze in verschiedenen Regionen verantwortlich. Es liegen folgende Daten vor.

	A	B	C	D	
1	<b>Monat</b>	<b>Region</b>	<b>Stadt</b>	<b>Umsatz</b>	
2	Januar	Nord	Kiel	500	
3	Januar	West	Krefeld	280	
4	Januar	Süd	Karlsruhe	700	
5	Januar	West	Köln	200	
6	Januar	Nord	Bremen	520	
7	Januar	Süd	Freiburg	480	
8	Februar	West	Köln	220	
9	Februar	Nord	Hamburg	450	
10	Februar	Süd	Karlsruhe	650	
11	Februar	West	Krefeld	300	
12	März	Nord	Bremen	600	
13	März	Süd	Karlsruhe	800	

Beispiel: Array oder Bereich mit Umsatzzahlen

Sie sollen die Daten nun nach Region gruppieren und summieren. Hierfür können Sie die folgende, einfache Formel verwenden:

**=GRUPPIERENNACH(B2:B13; D2:D13; SUMME)**

- **B2:B8:** row\_fields = Namen der Regionen, nach denen gruppiert werden soll.
- **C2:C8:** values = Werte, die je Region summiert werden sollen.
- **SUMME:** function = ist die Aggregatfunktion, die in der Gruppierung angewendet werden soll.

F2			=GRUPPIERENNACH(B2:B13;D2:D13;SUMME)				
	A	B	C	D	E	F	G
1	Monat	Region	Stadt	Umsatz		Region	Summe
2	Januar	Nord	Kiel	500		Nord	2.070
3	Januar	West	Krefeld	280		Süd	2.630
4	Januar	Süd	Karlsruhe	700		West	1.000
5	Januar	West	Köln	200		Gesamt	5.700
6	Januar	Nord	Bremen	520			
7	Januar	Süd	Freiburg	480			
8	Februar	West	Köln	220			
9	Februar	Nord	Hamburg	450			
10	Februar	Süd	Karlsruhe	650			
11	Februar	West	Krefeld	300			
12	März	Nord	Bremen	600			
13	März	Süd	Karlsruhe	800			

Ergebnis der Datenauswertung mit GRUPPERENNACH()

Diese einfachste Anwendung der GRUPPERENNACH()-Funktion summiert alle Werte in der Umsatzspalte, basierend auf den eindeutigen Werten in der Spalte mit den Regionen. Dies geschieht automatisch und ohne manuelle Sortierung oder Filterung.

## Weitere Parameter der Funktion GRUPPIERENNACH()

Nun wird die Ausgabe Schritt für Schritt durch die weiteren möglichen Einstellungen zur Funktion **GRUPPIERENNACH()** erläutert.

## Spaltenüberschriften anzeigen

Mit dem Parameter **field\_headers** werden bei der Ausgabe die Spaltenüberschriften übernommen. Der Datenbereich hat also Spaltenüberschriften, und diese sollen bei der Ausgabe angezeigt werden. Der Parameter `field_headers` erhält deshalb den Wert 3.

Die Funktion wird erweitert zu:

=GRUPPIERENNACH(B1:B13; D1:D13; SUMME; 3)

G1 : $\times$ $\checkmark$ $f_x$ $=GRUPPIERENNACH(B1:B13;D1:D13;SUMME;3)$								
	A	B	C	D	E	F	G	H
1	Monat	Region	Stadt	Umsatz		mit Header	Region	Umsatz
2	Januar	Nord	Kiel	500			Nord	2070
3	Januar	West	Krefeld	280			Süd	2630
4	Januar	Süd	Karlsruhe	700			West	1000
5	Januar	West	Köln	200			Gesamt	5700
6	Januar	Nord	Bremen	520				
7	Januar	Süd	Freiburg	480				
8	Februar	West	Köln	220				
9	Februar	Nord	Hamburg	450				
10	Februar	Süd	Karlsruhe	650				
11	Februar	West	Krefeld	300				
12	März	Nord	Bremen	600				
13	März	Süd	Karlsruhe	800				

Die Spaltenüberschriften (Header) werden angezeigt

## Ergebniszeilen anzeigen

Im nächsten Schritt soll keine Gesamtsumme angezeigt werden. Der Parameter **total\_depth** erhält deshalb den Wert .

Die Funktion wird nochmals erweitert zu:

`=GRUPPIERENNACH(B1:B13; D1:D13; SUMME; 3; )`

G1 : $\times$ $\checkmark$ $f_x$ $=GRUPPIERENNACH(B1:B13;D1:D13;SUMME;3;)$								
	A	B	C	D	E	F	G	H
1	Monat	Region	Stadt	Umsatz		mit Header	Region	Umsatz
2	Januar	Nord	Kiel	500		ohne Gesamt	Nord	2070
3	Januar	West	Krefeld	280			Süd	2630
4	Januar	Süd	Karlsruhe	700			West	1000
5	Januar	West	Köln	200				
6	Januar	Nord	Bremen	520				
7	Januar	Süd	Freiburg	480				
8	Februar	West	Köln	220				
9	Februar	Nord	Hamburg	450				
10	Februar	Süd	Karlsruhe	650				
11	Februar	West	Krefeld	300				
12	März	Nord	Bremen	600				
13	März	Süd	Karlsruhe	800				

Ausgabe ohne Gesamtsumme

## Ausgabe sortieren

Die Ausgabe der gruppierten Daten soll nun nach der Spalte **Region**, alphabetisch absteigend sortiert werden. Dazu wird dem Parameter **sort\_order** der Wert **-1** zugewiesen. Die Regionen werden in der Folge von Z bis A alphabetisch sortiert.

Die Funktion wird erweitert zu:

=GRUPPIERENNACH(B1:B13; D1:D13; SUMME; 3; 0; **-1**)

G1 :    =GRUPPIERENNACH(B1:B13;D1:D13;SUMME;3;0;-1)								
	A	B	C	D	E	F	G	H
1	Monat	Region	Stadt	Umsatz		mit Header	Region	Umsatz
2	Januar	Nord	Kiel	500		ohne Gesamt	West	1000
3	Januar	West	Krefeld	280		mit Sortierung	Süd	2630
4	Januar	Süd	Karlsruhe	700			Nord	2070
5	Januar	West	Köln	200				
6	Januar	Nord	Bremen	520				
7	Januar	Süd	Freiburg	480				
8	Februar	West	Köln	220				
9	Februar	Nord	Hamburg	450				
10	Februar	Süd	Karlsruhe	650				
11	Februar	West	Krefeld	300				
12	März	Nord	Bremen	600				
13	März	Süd	Karlsruhe	800				

*Sortierung der Gruppen (Region)*

Im nächsten Beispiel soll die Spalte „Stadt“ in die Ausgabe einbezogen werden. Die Sortierung der Ausgabe soll nach dem Kriterium „Stadt“ von Z bis A erfolgen.

Deshalb lautet der Parameter **-2**. Es wird die zweite Header-Spalte für die Sortierung genutzt, und diese soll alphabetisch von Z bis A sortiert werden. Die Funktion lautet entsprechend:

=GRUPPIERENNACH(B1:C13; D1:D13; SUMME; 3; 0; **-2**)



G1          =GRUPPIERENNACH(B1:C13;D1:D13;SUMME;3;0;-2)									
	A	B	C	D	E	F	G	H	I
1	Monat	Region	Stadt	Umsatz		mit Header	Region	Stadt	Umsatz
2	Januar	Nord	Kiel	500		ohne Gesamt	Nord	Kiel	500
3	Januar	West	Krefeld	280		mit Sortierung	Nord	Hamburg	450
4	Januar	Süd	Karlsruhe	700			Nord	Bremen	1120
5	Januar	West	Köln	200			Süd	Karlsruhe	2150
6	Januar	Nord	Bremen	520			Süd	Freiburg	480
7	Januar	Süd	Freiburg	480			West	Krefeld	580
8	Februar	West	Köln	220			West	Köln	420
9	Februar	Nord	Hamburg	450					
10	Februar	Süd	Karlsruhe	650					
11	Februar	West	Krefeld	300					
12	März	Nord	Bremen	600					
13	März	Süd	Karlsruhe	800					

Sortierung innerhalb der Gruppe „Region“ nach der Untergruppe „Stadt“

Das Ergebnis zeigt, dass die Sortierung in der Untergruppe „Stadt“ nur innerhalb der übergeordneten Gruppe „Region“ erfolgt. Die Region wird alphabetisch sortiert und in der Region „Nord“ werden die Städte alphabetisch von Z bis A sortiert.

## Filter einsetzen

Mit dem Parameter **filter\_array** können Sie Kriterien definieren, nach denen Zeilen (Datensätze) aus der Ausgabe herausgefiltert werden. Im Beispiel soll das Ergebnis aus „Kiel“ nicht berücksichtigt werden.

Dann lautet die erweiterte Funktion:

=GRUPPIERENNACH(B1:C13; D1:D13; SUMME; 3; 0; -2; **C1:C13<>"Kiel"**)

G1 : $\times$ $\checkmark$ $f_x$ =GRUPPIERENNACH(B1:C13;D1:D13;SUMME;3;0;-2;C1:C13<>"Kiel")									
	A	B	C	D	E	F	G	H	I
1	Monat	Region	Stadt	Umsatz		mit Header	Region	Stadt	Umsatz
2	Januar	Nord	Kiel	500		ohne Gesamt	Nord	Hamburg	450
3	Januar	West	Krefeld	280		mit Sortierung	Nord	Bremen	1120
4	Januar	Süd	Karlsruhe	700		mit Filterung	Süd	Karlsruhe	2150
5	Januar	West	Köln	200			Süd	Freiburg	480
6	Januar	Nord	Bremen	520			West	Krefeld	580
7	Januar	Süd	Freiburg	480			West	Köln	420
8	Februar	West	Köln	220					
9	Februar	Nord	Hamburg	450					
10	Februar	Süd	Karlsruhe	650					
11	Februar	West	Krefeld	300					
12	März	Nord	Bremen	600					
13	März	Süd	Karlsruhe	800					

Ergebnis von GRUPPIERENNACH() bei Filterung


## Hierarchie der Gruppierung beachten

Da im Beispiel zwei Kriterien für die Gruppierung genutzt werden (Region und Stadt), können Sie mit dem Parameter **field\_relationship** einstellen, ob bei der Sortierung die Hierarchie der Gruppierung beachtet werden soll oder nicht.

Bei der Standardeinstellung wird die Gruppe „Region“ ausgegeben und sortiert und dann – innerhalb dieser Gruppe – die Untergruppe „Stadt“ ausgegeben und sortiert.

Soll diese Hierarchie nicht beachtet werden, dann wählen Sie für den Parameter **field\_relationship** den Wert **1**, wie im folgenden Beispiel. Die Funktion lautet:

=GRUPPIERENNACH(B1:C13; D1:D13; SUMME; 3; 0; -2; C1:C13<>"Kiel"; **1**)

G1 :  =GRUPPIERENNACH(B1:C13;D1:D13;SUMME;3;0;-2;C1:C13<>"Kiel";1)

	A	B	C	D	E	F	G	H	I
1	Monat	Region	Stadt	Umsatz		mit Header	Region	Stadt	Umsatz
2	Januar	Nord	Kiel	500		ohne Gesamt	West	Krefeld	580
3	Januar	West	Krefeld	280		mit Sortierung	West	Köln	420
4	Januar	Süd	Karlsruhe	700		mit Filterung	Süd	Karlsruhe	2150
5	Januar	West	Köln	200		ohne Hierarchie	Nord	Hamburg	450
6	Januar	Nord	Bremen	520			Süd	Freiburg	480
7	Januar	Süd	Freiburg	480			Nord	Bremen	1120
8	Februar	West	Köln	220					
9	Februar	Nord	Hamburg	450					
10	Februar	Süd	Karlsruhe	650					
11	Februar	West	Krefeld	300					
12	März	Nord	Bremen	600					
13	März	Süd	Karlsruhe	800					

Keine Gruppenhierarchie bei der Ausgabe mit GRUPPIERENNACH()

Die Sortierung der Ausgabe erfolgt nur nach dem Kriterium „Stadt“, in der angegebenen Reihenfolge alphabetisch von Z bis A. Die übergeordnete Gruppierung „Region“ wird nicht berücksichtigt.

## Warum ist GRUPPIERENNACH() ein Gamechanger?

- **Effizienz:** Kein langwieriges Setup von Pivot-Tabellen mehr.
- **Flexibilität:** Funktioniert perfekt mit dynamischen Daten und Arrays.
- **Automatisierung:** Ideal für Dashboards und Berichte, die sich mit den zugrundeliegenden Daten automatisch aktualisieren.
- **Leicht verständlich:** Die Syntax ist intuitiv und schnell anzuwenden.
- **Mit dynamischen Arrays kombinieren:** Nutzen Sie GRUPPIERENNACH() mit Funktionen wie FILTER() oder SORTIEREN() für noch mächtigere Analysen.
- **Benutzerdefinierte Aggregationen:** Wenn Standardfunktionen wie SUMME oder ANZAHL nicht ausreichen, können Sie **benutzerdefinierte LAMBDA()-Funktionen** integrieren.

Die neue GRUPPIERENNACH()-Funktion spart Zeit, reduziert Fehler und bringt die Datenanalyse auf ein neues Niveau. Sie eignet sich perfekt für alle, die große Datenmengen schnell und übersichtlich zusammenfassen möchten.

Probieren Sie es aus und stellen Sie die Parameter so ein, bis die Ausgabe Ihren Erwartungen entspricht. Es ist wirklich sehr einfach!

## Mit SPALTENWAHL() gezielt Spalten in einer Excel-Tabelle auswählen

Mit der Excel-Funktion SPALTENWAHL() wählen Sie aus umfangreichen Tabellen nur die gewünschten Spalten aus. Und in Kombination mit FILTER() oder SEQUENZ() können Sie die Auswahl verfeinern.

Zuletzt geändert am 18.03.2026



Mit der neuen Funktion **SPALTENWAHL()** in Excel können Sie gezielt bestimmte Spalten aus einem Bereich auswählen, ohne umständliche Hilfskonstruktionen zu basteln.

In diesem Beitrag erfahren Sie, wie die Funktion SPALTENWAHL() funktioniert, wo sie Ihnen in der Praxis echte Arbeit abnimmt – und wie Sie die Funktion clever kombinieren.

### Was macht SPALTENWAHL()?

SPALTENWAHL() erlaubt es Ihnen, aus einem Datenbereich ganz einfach nur bestimmte Spalten auszuwählen.

Die Struktur ist dabei denkbar einfach:

**=SPALTENWAHL(Bereich; Spalte1; [Spalte2] ...)**

- **Bereich:** Ihre Ausgangstabelle
- **Spalte1:** Die Nummer der Spalte, die Sie herausfiltern möchten
- **Spalte2 ...** (optional): Die Nummern weiterer Spalten, die Sie herausfiltern möchten

Dabei beginnt die Zählung natürlich bei **1**. Das ist die **erste linke Spalte** des angegebenen Bereichs.

## Praxisbeispiel

Sie haben in der Personalabteilung eine Liste der Mitarbeitenden mit den Angaben wie in der folgenden Abbildung.

	A	B	C	D	E
1	<b>Mitarbeitende</b>	<b>Abteilung</b>	<b>Standort</b>	<b>Eintrittsdatum</b>	<b>Gehalt (€)</b>
2	Anna Müller	Buchhaltung	Würzburg	01.01.2020	45.000
3	Bernd Becker	Einkauf	Schweinfurt	15.06.2019	48.000
4	Carla Schmitt	IT	Würzburg	01.09.2021	52.000
5	Daniel König	Vertrieb	Bamberg	10.10.2018	50.000
6	Eva Schneider	Marketing	Nürnberg	20.03.2022	47.500
7					

Beispiel: Tabelle mit mehreren Spalten und unterschiedlichen Daten

Sie möchten daraus nur den Namen und den Standort anzeigen? Kein Problem!

Mit folgender Formel holen Sie sich genau diese beiden Spalten:

**=SPALTENWAHL(A1:E6; 1; 3)**

Das Ergebnis ist eine übersichtliche Liste – nur mit den gewünschten Informationen.

G1          **=SPALTENWAHL(A1:E6; 1; 3)**

	A	B	C	D	E	F	G	H
1	<b>Mitarbeitende</b>	<b>Abteilung</b>	<b>Standort</b>	<b>Eintrittsdatum</b>	<b>Gehalt (€)</b>		<b>Mitarbeitende</b>	<b>Standort</b>
2	Anna Müller	Buchhaltung	Würzburg	01.01.2020	45.000		Anna Müller	Würzburg
3	Bernd Becker	Einkauf	Schweinfurt	15.06.2019	48.000		Bernd Becker	Schweinfurt
4	Carla Schmitt	IT	Würzburg	01.09.2021	52.000		Carla Schmitt	Würzburg
5	Daniel König	Vertrieb	Bamberg	10.10.2018	50.000		Daniel König	Bamberg
6	Eva Schneider	Marketing	Nürnberg	20.03.2022	47.500		Eva Schneider	Nürnberg
7								

### Funktion SPALTENWAHL() in Excel

Sie müssen sich also nicht mehr mit INDEX() oder anderen komplexen Formelkonstruktionen abmühen – eine einzige Standardfunktion reicht aus.

## SPALTENWAHL() kombiniert mit FILTER()

Wenn Sie nicht nur bestimmte Spalten auswählen, sondern auch bestimmte Zeilen filtern wollen, kombinieren Sie die beiden Funktionen **SPALTENWAHL()** und **FILTER()**.

**Beispiel:** Sie wollen nur die Mitarbeitenden aus Würzburg anzeigen – und möchten nur den Namen und das Eintrittsdatum sehen.

Hier die passende Formel:

**=SPALTENWAHL(FILTER(A2:E7; C2:C7="Würzburg"); 1; 4)**

	A	B	C	D	E	F	G	H
1	Mitarbeitende	Abteilung	Standort	Eintrittsdatum	Gehalt (€)			
2	Anna Müller	Buchhaltung	Würzburg	01.01.2020	45.000		Anna Müller	01.01.2020
3	Bernd Becker	Einkauf	Schweinfurt	15.06.2019	48.000		Carla Schmitt	01.09.2021
4	Carla Schmitt	IT	Würzburg	01.09.2021	52.000			
5	Daniel König	Vertrieb	Bamberg	10.10.2018	50.000			
6	Eva Schneider	Marketing	Nürnberg	20.03.2022	47.500			
7								

Funktionen SPALTENWAHL() und FILTER() kombiniert

**FILTER()** sucht alle Zeilen, in denen der Standort "Würzburg" ist. Das Ergebnis ist der betrachtete Bereich für die Funktion **SPALTENWAHL()**. Sie reduziert das Ergebnis auf die erste und vierte Spalte (Name und Eintrittsdatum).

## Dynamische Auswahl von Spalten mit SEQUENZ()

Manchmal möchten Sie nicht einzelne Spalten fix angeben, sondern dynamisch die ersten x Spalten auswählen. Hier kommt die Funktion **SEQUENZ()** ins Spiel, die zusammen mit **SPALTENWAHL()** eine starke Kombination bildet.

**Beispiel:** Sie möchten immer die ersten drei Spalten Ihrer Liste anzeigen? Die dafür geeignete Formel lautet:

**=SPALTENWAHL(A2:E7; SEQUENZ(3))**

SEQUENZ(3) erzeugt automatisch die Werte 1, 2 und 3. SPALTENWAHL() nutzt diese Werte, um die entsprechenden Spalten herauszufiltern.

Das Ergebnis zeigt für das Beispiel mit den Personaldaten genau die Spalten Name, Abteilung und Standort.

G1 <b>=SPALTENWAHL(A1:E6; SEQUENZ(3))</b>									
	A	B	C	D	E	F	G	H	I
1	Mitarbeitende	Abteilung	Standort	Eintrittsdatum	Gehalt (€)		Mitarbeitende	Abteilung	Standort
2	Anna Müller	Buchhaltung	Würzburg	01.01.2020	45.000		Anna Müller	Buchhaltung	Würzburg
3	Bernd Becker	Einkauf	Schweinfurt	15.06.2019	48.000		Bernd Becker	Einkauf	Schweinfurt
4	Carla Schmitt	IT	Würzburg	01.09.2021	52.000		Carla Schmitt	IT	Würzburg
5	Daniel König	Vertrieb	Bamberg	10.10.2018	50.000		Daniel König	Vertrieb	Bamberg
6	Eva Schneider	Marketing	Nürnberg	20.03.2022	47.500		Eva Schneider	Marketing	Nürnberg
7									

Funktionen SPALTENWAHL() und SEQUENZ() kombiniert

Ein großer **Vorteil**: Ändern Sie einfach die Zahl in SEQUENZ(), passen sich die ausgewählten Spalten automatisch an.

Die Funktion SPALTENWAHL() macht Excel noch flexibler. Mit SPALTENWAHL() sparen Sie Zeit, vereinfachen Ihre Formeln und erstellen schneller genau die Auswertungen, die Sie brauchen.

In Kombination mit Funktionen wie FILTER() und SEQUENZ() wird aus einer kleinen Funktion ein mächtiges Werkzeug für Ihre tägliche Excel-Praxis.



Den Excel-Tipp zur entsprechenden **Zeilenwahl** finden Sie hier: **So wählen Sie mit ZEILENWAHL() einzelne Zeilen aus.**

## So wählen Sie mit ZEILENWAHL() einzelne Zeilen einer Excel-Tabelle aus

Mit der einfachen Funktion ZEILENWAHL() legen Sie fest, welche Zeilen Sie aus einer Liste oder Tabelle auswählen und ausgeben wollen. So gehen Sie vor – auch in Kombination mit anderen Excel-Funktionen.

Zuletzt geändert am 18.03.2026



Mit der Funktion **ZEILENWAHL()** können Sie gezielt bestimmte **Zeilen** aus einer bestehenden Tabelle **auswählen** – ohne komplizierte Hilfsspalten oder Filtereinstellungen. Gerade in der betrieblichen Praxis eröffnet diese Funktion viele Möglichkeiten für dynamische Auswertungen.

In diesem Beitrag erläutern wir, wie ZEILENWAHL() funktioniert, und stellen Ihnen ein Beispiel aus dem Bereich Mitarbeiterverwaltung vor.

### Was macht ZEILENWAHL()?

Die Excel-Funktion **ZEILENWAHL()** erstellt aus einem Zellbereich oder einer Tabelle eine neue Zusammenstellung – und zwar basierend auf den Zeilennummern, die Sie angeben. Dabei bleiben die Spalten und die Struktur der Tabelle erhalten.

Die Funktion ZEILENWAHL() hat die folgende Syntax:

**=ZEILENWAHL(Bereich; Zeile1; [Zeile2] ...)**

- **Bereich:** Der Bereich oder die Tabelle, aus denen Zeilen ausgewählt werden sollen
- **Zeile1:** Die Nummer der gewünschten Zeile; beginnend bei 1 für die erste Zeile mit den Daten
- **Zeile2 ... (optional):** Die Nummer weiterer gewünschter Zeilen



**Wichtig:** Die Zählung in ZEILENWAHL() startet immer relativ zum angegebenen Bereich. Das heißt: Im folgenden Beispiel ist A2 die erste Zeile – nicht A1, in der sich der Spaltentitel befindet.

## Beispiel aus der Praxis: Auswahl einzelner Personen aus der Mitarbeiterliste

Sie haben eine Liste aller Mitarbeitenden Ihres Unternehmens und möchten gezielt nur diejenigen aus den Abteilungen IT und Buchhaltung herausfiltern – etwa für eine gezielte Schulungseinladung.

Die Beispieltabelle (Blatt „Mitarbeitende“) sieht so aus:

	A	B	C
1	<b>Mitarbeitende</b>	<b>Abteilung</b>	<b>Standort</b>
2	Anna Müller	Buchhaltung	Würzburg
3	Bernd Becker	Einkauf	Schweinfurt
4	Carla Schmitt	IT	Würzburg
5	Daniel Roth	Vertrieb	Nürnberg
6	Elena König	IT	Schweinfurt
7	Frank Wagner	Buchhaltung	Würzburg
8			

*Beispiel: Liste mit den Beschäftigten im Unternehmen*

## Anwendung der ZEILENWAHL()-Funktion

Sie wissen: Die Mitarbeitenden aus IT und Buchhaltung befinden sich in den Zeilen 1, 3, 5 und 6 (bezogen auf die Daten – nicht auf die gesamte Tabelle inklusive Kopfzeile).

Die passende Formel zum Herausfiltern dieser Zeilen lautet daher wie folgt:

**=ZEILENWAHL(A2:C7; 1; 3; 5; 6)**

Excel zeigt eine neue Liste nur mit den ausgewählten Personen – ohne Filtern oder Kopieren!

E1 $\times$ $\checkmark$ $f_x$ <b>=ZEILENWAHL(A2:C7;1;3;5;6)</b>						
	A	B	C	D	E	F
1	Mitarbeitende	Abteilung	Standort		Anna Müller	Buchhaltung
2	Anna Müller	Buchhaltung	Würzburg		Carla Schmitt	IT
3	Bernd Becker	Einkauf	Schweinfurt		Elena König	IT
4	Carla Schmitt	IT	Würzburg		Frank Wagner	Buchhaltung
5	Daniel Roth	Vertrieb	Nürnberg			
6	Elena König	IT	Schweinfurt			
7	Frank Wagner	Buchhaltung	Würzburg			
8						

Funktion ZEILENWAHL() in Excel

## Vorteile von ZEILENWAHL()

Sie erkennen schnell die Vorteile der Funktion:

- Gezielte Selektionen: Ideal für individuelle Berichte oder Einladungslisten.
- Dynamische Auswertungen: In Kombination mit anderen Funktionen wie FILTER() können Sie Dashboards erstellen.
- Lesbarkeit: Ihre Formeln bleiben kurz und verständlich.

## ZEILENWAHL() kombiniert mit FILTER()

Noch leistungsfähiger wird ZEILENWAHL(), wenn sie mit der Funktion FILTER() kombiniert wird.

**Beispiel:** Sie möchten zunächst alle Mitarbeitenden am Standort Würzburg herausfiltern – und anschließend nur bestimmte Personen daraus weiterverarbeiten.

### Schritt 1: Auswahl mit FILTER()

Mit der Funktion FILTER() können Sie aus der gesamten Mitarbeiterliste gezielt die Mitarbeitenden aus Würzburg extrahieren. Die Funktion lautet:

**=FILTER(A2:C7; C2:C7="Würzburg")**

So wählen Sie mit ZEILENWAHL() einzelne Zeilen einer Excel-Tabelle aus

E1 <span>fx</span> =FILTER(A2:C7;C2:C7="Würzburg")						
	A	B	C	D	E	F
1	Mitarbeitende	Abteilung	Standort		Anna Müller	Buchhaltung Würzburg
2	Anna Müller	Buchhaltung	Würzburg		Carla Schmitt	IT Würzburg
3	Bernd Becker	Einkauf	Schweinfurt		Frank Wagner	Buchhaltung Würzburg
4	Carla Schmitt	IT	Würzburg			
5	Daniel Roth	Vertrieb	Nürnberg			
6	Elena König	IT	Schweinfurt			
7	Frank Wagner	Buchhaltung	Würzburg			
8						

Die Funktion FILTER() in Excel

## Schritt 2: Weitere Auswahl mit ZEILENWAHL()

Nun möchten Sie aus dieser neuen, gefilterten Liste nur die erste und dritte Zeile auswählen; im Beispiel sind das Anna Müller und Frank Wagner.

Dafür kombinieren Sie ZEILENWAHL() mit FILTER():

**=ZEILENWAHL(FILTER(A2:C7; C2:C7="Würzburg"); 1; 3)**

E1 <span>fx</span> =ZEILENWAHL(FILTER(A2:C7;C2:C7="Würzburg");1;3)						
	A	B	C	D	E	F
1	Mitarbeitende	Abteilung	Standort		Anna Müller	Buchhaltung Würzburg
2	Anna Müller	Buchhaltung	Würzburg		Frank Wagner	Buchhaltung Würzburg
3	Bernd Becker	Einkauf	Schweinfurt			
4	Carla Schmitt	IT	Würzburg			
5	Daniel Roth	Vertrieb	Nürnberg			
6	Elena König	IT	Schweinfurt			
7	Frank Wagner	Buchhaltung	Würzburg			
8						

Kombination der Excel-Funktionen ZEILENWAHL() und FILTER()

Die Funktion **FILTER()** reduziert die Tabelle auf Einträge, die einem bestimmten Kriterium entsprechen. Die Funktion **ZEILENWAHL()** erlaubt danach eine gezielte Auswahl einzelner Einträge, etwa die erste und dritte gefundene Person.

Das Ganze erfolgt dynamisch: Wenn sich die Ausgangsdaten ändern, aktualisieren sich auch die Filter- und Auswahlresultate automatisch.

Diese Technik eignet sich besonders gut für Berichte, in denen zum Beispiel nur die besten Kunden, die wichtigsten Projekte oder ausgewählte Standorte präsentiert werden sollen.

## Dynamische Auswahl mit SEQUENZ() und ZEILENWAHL()

Nun möchten Sie nicht manuell bestimmte Zeilen angeben, sondern eine dynamische Anzahl an Zeilen auswählen – zum Beispiel immer die ersten drei Mitarbeitenden aus einer Liste.

Hier kommt die Funktion **SEQUENZ()** ins Spiel. Mit SEQUENZ() können Sie eine fortlaufende Zahlenreihe erstellen, die Sie dann in ZEILENWAHL() übergeben.

**=ZEILENWAHL(A2:C7; SEQUENZ(3))**

Was passiert hier?

- **A2:C7**: Der Datenbereich für die Auswahl
- **SEQUENZ(3)**: Erzeugt die Zeilennummern 1, 2 und 3
- **ZEILENWAHL()**: Wählt die ersten drei Zeilen entsprechend der erzeugten Sequenz aus

	A	B	C	D	E	F	G
1	arbeitende	Abteilung	Standort		Anna Müller	Buchhaltung	Würzburg
2	na Müller	Buchhaltung	Würzburg		Bernd Becker	Einkauf	Schweinfurt
3	nd Becker	Einkauf	Schweinfurt		Carla Schmitt	IT	Würzburg
4	la Schmitt	IT	Würzburg				
5	niel Roth	Vertrieb	Nürnberg				
6	ena König	IT	Schweinfurt				
7	nk Wagner	Buchhaltung	Würzburg				
8							

Kombination der Excel-Funktionen ZEILENWAHL() und SEQUENZ()



Den Excel-Tipp zur entsprechenden **Spaltenwahl** finden Sie hier: **Mit SPALTENWAHL() gezielt Spalten auswählen.**

## Mit SVERWEIS() Einträge in Excel-Tabellen suchen

Mit der Excel-Funktion SVERWEIS() finden Sie in einer Tabelle mit Daten genau die Einträge, die zu Ihrem Suchkriterium passen. So können Sie Datensätze einfach und schnell auswerten.

Zuletzt geändert am 18.03.2026



Die Funktion **SVERWEIS()** ist eine der wichtigsten Funktionen in Excel. Mit SVERWEIS() können Sie zum Beispiel auf einfache Art und Weise solche Daten aus einer Tabelle an eine andere Stelle übertragen, die einem festgelegten Suchkriterium entsprechen.

### Syntax für die Funktion SVERWEIS()

Die Syntax für die Funktion SVERWEIS() lautet:

**=SVERWEIS(Suchkriterium; Bereich; Spalte; [Bereich\_Verweis])**

Mit welchen Argumenten wird die Funktion **SVERWEIS()** gesteuert?

- **Suchkriterium:** Ist der Wert, den Sie in der Datentabelle nachschlagen müssen.
- **Bereich:** Ist der Bereich, in dem Sie bestimmte Daten suchen. Sie sollten die Datentabelle immer als absoluten Bezug mit Dollarzeichen erfassen, damit sich der Matrixbereich beim Kopieren nicht verschiebt.
- **Spalte:** In der Datentabelle die Nummer der Spalte, die den Rückgabewert enthält.
- **Bereich\_Verweis** (optional): Dieses Argument gibt an, ob eine genaue Übereinstimmung (FALSCH) oder ungefähre Übereinstimmung (WAHR = Standard) zutreffen soll.

## Funktionsweise von SVERWEIS() für das Praxisbeispiel

In der folgenden Abbildung sehen Sie in den Spalten A:C einen Tabellenbereich, in den durch die Eingabe der Produktnummer der Produktname und der Preis aus der Tabelle mit den Daten in den Spalten E:F (roter Kasten) automatisch übertragen werden sollen.

B2							
	A	B	C	D	E	F	G
1	Produktnummer	Produkt	Preis		Produktnummer	Produkt	Preis
2	103				101	Produkt A	100,00
3					102	Produkt B	200,00
4					103	Produkt C	150,00
5					104	Produkt D	50,00
6							
7							

Praxisbeispiel für SVERWEIS() mit der auszuwertenden Datentabelle (roter Kasten)

Um den Produktnamen automatisch zu übertragen, erfassen Sie die folgende Formel in der Zelle B2 und bestätigen Sie Ihre Eingabe durch die Enter-Taste:

**=SVERWEIS(A2; \$E\$2:\$G\$5; 2; FALSCH)**

Mit dieser Formel suchen Sie den im Bereich E2:G5 den Produktnamen zur Produktnummer 103.

Das Ergebnis ist „Produkt C“, das in die Zelle B2 übertragen wird. Würden Sie die Produktnummer in der Zelle A2 ändern, dann wird automatisch der jeweilige Produktname entsprechend angepasst.

B2							
							=SVERWEIS(A2;\$E\$2:\$G\$5;2;FALSCH)
	A	B	C	D	E	F	G
1	Produktnummer	Produkt	Preis		Produktnummer	Produkt	Preis
2	103	Produkt C			101	Produkt A	100,00
3					102	Produkt B	200,00
4					103	Produkt C	150,00
5					104	Produkt D	50,00
6							

Auswertung einer Datentabelle mit SVERWEIS()

**Merke****Gesucht wird in der ersten Spalte der Datentabelle**

Das Suchkriterium – im Beispiel die Produktnummer in Zelle A2 – muss immer in der ersten Spalte der Datentabelle enthalten sein!

Die Funktion SVERWEIS() sucht in der Datentabelle (\$E\$2:\$G\$5) in der ersten Spalte den Suchbegriff (A = Produktnummer = 103) **senkrecht** nach unten. Das **S** in SVERWEIS() steht für senkrecht!

Findet die Funktion SVERWEIS() in der ersten Spalte der Matrix exakt den gesuchten Wert, dann hält sie die Suche an und wandert so viele Spalten nach rechts, wie Sie bei Spaltennummer angegeben haben.

Der Wert, der dann in der entsprechenden Zelle steht, wird als Ergebnis zurückgegeben.

E	F	G
Produktnummer	Produkt	Preis
101	Produkt A	100,00
102	Produkt B	200,00
103	Produkt C	150,00
104	Produkt D	50,00

*So wertet Excel die Datentabelle mit SVERWEIS() aus*

Um jetzt auch noch den Preis in der Zelle C2 zu übertragen, müssen Sie die SVERWEIS()-Formel nur in der Spaltennummer anpassen, da der Preis in der dritten Spalte der Datentabelle steht.

`=SVERWEIS(A2; $E$2:$G$5; 3; FALSCH)`

C2 $\downarrow$ $\times$ $\checkmark$ $f_x$ =SVERWEIS(A2;\$E\$2:\$G\$5;3;FALSCH)							
	A	B	C	D	E	F	G
1	Produktnummer	Produkt	Preis		Produktnummer	Produkt	Preis
2	103	Produkt C	150		101	Produkt A	100,00
3					102	Produkt B	200,00
4					103	Produkt C	150,00
5					104	Produkt D	50,00

Auswertung einer Datentabelle mit SVERWEIS() – Auswahl der Angabe in der dritten Spalte

## Ein Beispiel für ungefähre Übereinstimmung (WAHR)

Die ungefähre Übereinstimmung (Argument **WAHR**) beim SVERWEIS() können Sie sich zunutze machen, wenn Sie Werte oder Inhalte staffelweise zuordnen wollen.

In der folgenden Abbildung sehen Sie in den Spalten E:F eine umsatzabhängige, gestaffelte Bonustabelle. Um die entsprechenden Bonussätze abhängig vom Umsatz mit dem SVERWEIS() zu übernehmen, können Sie den SVERWEIS() wie in der folgenden Abbildung mit dem Argument WAHR verwenden.

B2 $\downarrow$ $\times$ $\checkmark$ $f_x$ =SVERWEIS(A2;\$E\$2:\$F\$6;2;WAHR)						
	A	B	C	D	E	F
1	Umsatz	Bonus			Umsatz	Bonus
2	70.000,00	5%			0,00	0%
3					50.000,00	5%
4					100.000,00	10%
5					150.000,00	15%
6					200.000,00	20%
7						

Auswertung einer Datentabelle mit SVERWEIS() und ungefährer Übereinstimmung beim Suchkriterium

Der SVERWEIS() geht mit dem **Argument WAHR** hier wie folgt vor.

- Zuerst wird der Suchbegriff (70.000,00) in der ersten Spalte der Datentabelle gesucht.
- Wird der Begriff exakt gefunden, dann wird die zweite Spalte in der Zeile, in welcher der Begriff gefunden wurde, als Ergebnis zurückgegeben.



- Wird der Suchbegriff nicht exakt gefunden, dann geht SVERWEIS() alle Einträge in der ersten Spalte der Datentabelle durch, bis er auf den Wert 100.000,00 stößt. Dieser Wert ist größer als 70.000,00.
- Anschließend springt der SVERWEIS() einen Eintrag zurück nach oben auf die 50.000,00 und gibt den Bonussatz hier in der zweiten Spalte zurück.

### Merke

#### **Datentabelle für das Argument WAHR nach dem Suchkriterium sortieren**

Damit Sie den SVERWEIS() mit ungefähre Übereinstimmung (WAHR) einsetzen können, müssen Sie die Datentabelle aufsteigend nach dem Suchkriterium (erste Spalte) sortieren.

## Excel-Funktion SVERWEIS() bei Groß- und Kleinschreibung

Bei der Auswertung von Daten mit der wichtigen Funktion SVERWEIS() werden Groß- und Kleinschreibung nicht berücksichtigt. Kommt es aber auf diese Schreibweise an, gehen Sie folgendermaßen vor.

Zuletzt geändert am 18.03.2026



**SVERWEIS()** ist eine der wichtigsten Funktionen in Excel überhaupt. Mit ihm können Sie Werte aus einer Liste gezielt zurückgeben, wenn das Suchkriterium mit dem Listeneintrag übereinstimmt.

**Aber:** Die Funktion SVERWEIS() kann bei Texten im Suchkriterium nicht zwischen Groß- und Kleinschreibung unterscheiden.

### Die Standardanwendung von SVERWEIS()

In der folgenden Abbildung sehen Sie ein Beispiel, in welchem der Name LEROY SANÉ gesucht wird.

Da SVERWEIS() nicht zwischen Groß- und Kleinschreibung unterscheidet, wird als Ergebnis der Wert 5 zurückgegeben. Richtig wäre der Wert 6, da der großgeschriebene Name erst eine Zeile tiefer in der Liste vorkommt.

F2    :    ✕    ✓    fx    =SVERWEIS(D2;A:B;2;FALSCH)						
	A	B	C	D	E	F
1	Name	Nummer		Suchbegriff		Ergebnis
2	Thomas Müller	1		LEROY SANÉ		5
3	Ilkay Gündoğan	2				
4	Kai Havertz	3				
5	Joshua Kimmich	4				
6	Leroy Sané	5				
7	LEROY SANÉ	6				
8	Serge Gnabry	7				
9	Antonio Rüdiger	8				
10	Marc-André ter Stegen	9				
11	Jamal Musiala	10				

So funktioniert SVERWEIS() standardmäßig

Das gleiche Ergebnis erhalten Sie bei der Funktion **XVERWEIS()**. Auch hier wird nicht zwischen Groß- und Kleinschreibung unterschieden.

F2    :    ✕    ✓    fx    =XVERWEIS(D2;A:A;B:B;"";0;1)						
	A	B	C	D	E	F
1	Name	Nummer		Suchbegriff		Ergebnis
2	Thomas Müller	1		LEROY SANÉ		5
3	Ilkay Gündoğan	2				
4	Kai Havertz	3				
5	Joshua Kimmich	4				
6	Leroy Sané	5				
7	LEROY SANÉ	6				
8	Serge Gnabry	7				
9	Antonio Rüdiger	8				
10	Marc-André ter Stegen	9				
11	Jamal Musiala	10				

XVERWEIS() unterscheidet keine Groß- und Kleinschreibung

## Die Alternative für SVERWEIS() bei Groß- und Kleinschreibung

Mit der folgenden Matrixformel können Sie die Funktion SVERWEIS() „simulieren“ und die Groß- und Kleinschreibung im Suchkriterium berücksichtigen:

**{=INDEX(B2:B11; VERGLEICH(WAHR; IDENTISCH(D2; A2:A11); 0))}**

Die Formel geht von den folgenden Annahmen aus:

- **Suchbegriff:** der gesucht werden soll: **D2**
- **Suchspalte:** Liste, in der nach dem Suchbegriff gesucht wird: **A2:A11**
- **Wertspalte:** Liste, aus welcher der passende Wert zurückgegeben werden soll: **B2:B11**

Erfassen Sie die Formel ohne die geschweiften Klammern. Bestätigen Sie die Formel dann aber nicht mit Enter, sondern mit der Tastenkombination **Strg + Umschalttaste + Enter**, da es sich um eine **Matrixformel** handelt. Durch die Tastenkombination werden automatisch die geschweiften Klammern am Anfang und am Ende eingefügt.

F2	{=INDEX(B2:B11; VERGLEICH(WAHR; IDENTISCH(D2; A2:A11); 0))}					
	A	B	C	D	E	F
1	Name	Nummer		Suchbegriff		Ergebnis
2	Thomas Müller	1		LEROY SANÉ		6
3	Ilkay Gündoğan	2				
4	Kai Havertz	3				
5	Joshua Kimmich	4				
6	Leroy Sané	5				
7	LEROY SANÉ	6				
8	Serge Gnabry	7				
9	Antonio Rüdiger	8				
10	Marc-André ter Stegen	9				
11	Jamal Musiala	10				

Liste auswerten mit Beachtung der Groß- und Kleinschreibung

## So funktioniert der „SVERWEIS()“ bei Groß- und Kleinschreibung

Zunächst wird der Suchbegriff in der Zelle D2 mit den Namen im Bereich A2:A11 mit der Funktion IDENTISCH() verglichen, ob diese Werte mit dem Suchbegriff identisch sind. Hierbei wird auch die Groß- und Kleinschreibung beachtet.

**{=INDEX(B2:B11; VERGLEICH(WAHR; IDENTISCH(D2; A2:A11); 0))}**

Durch diesen Vergleich ergibt sich eine Liste aus WAHR- und FALSCH-Werten. Stimmen die Namen unter Beachtung der Groß- und Kleinschreibung überein, dann wird der Wert WAHR, sonst der Wert FALSCH zurückgegeben.

Im Beispiel oben würde die Liste wie folgt aussehen: FALSCH, FALSCH, FALSCH, FALSCH, FALSCH, WAHR, FALSCH, FALSCH, FALSCH, FALSCH. Der sechste Eintrag ist WAHR, da hier der Name unter Beachtung der Groß- und Kleinschreibung übereinstimmt.

Diese Liste wird nun an die Funktion VERGLEICH() weitergeleitet. Hier wird jeder einzelne Wert der Liste mit dem Wert WAHR verglichen.

**{=INDEX(B2:B11; VERGLEICH(WAHR; IDENTISCH(D2; A2:A11); 0))}**

Es wird die Nummer des Listeneintrags zurückgegeben, bei welchem der Vergleich übereinstimmt. Im Beispiel stimmt der sechste Eintrag in der Liste mit dem Wert WAHR überein.

Der Parameter 0 in der Funktion VERGLEICH() gibt an, dass nach identischen Einträgen gesucht werden soll.

Somit wird der Wert 6 als Ergebnis an die Funktion INDEX() als zweites Argument weitergegeben.

**{=INDEX(B2:B11; VERGLEICH(WAHR; IDENTISCH(D2; A2:A11); 0))}**

Die Funktion INDEX() sucht in der Liste B2:B11 den sechsten Eintrag und liefert den Wert dieser Zelle als Ergebnis zurück. Dies ist im Beispiel der gesuchte Wert 6, der in der Spalte B neben dem Namen LEROY SANÉ steht.

## Mit WVERWEIS() Excel-Tabellen durchsuchen

Mit der Excel-Funktion WVERWEIS() finden Sie in einer Tabelle mit Daten die Einträge, die zu Ihrem Suchkriterium passen. Der WVERWEIS() sucht – anders als der SVERWEIS() – waagerecht, von links nach rechts. Welche Funktion eingesetzt wird, hängt von der Anordnung der Daten ab.

Zuletzt geändert am 18.03.2026



Mit der **Funktion SVERWEIS()** in Excel können Sie Daten aus einer Tabelle auslesen und an eine andere Stelle übertragen. Neben SVERWEIS() gibt es auch die Funktion WVERWEIS(), die zwar nicht so bekannt, aber ebenfalls sehr nützlich für Ihre tägliche Arbeit mit Excel ist.

Der Unterschied zwischen beiden Funktionen besteht lediglich in der Anordnung der Matrix (Tabelle), aus der Daten ausgewählt werden. Der Unterschied:

- Beim **SVERWEIS** (das S steht für senkrecht) sind die Daten vertikal angeordnet und werden von oben nach unten durchsucht.
- Beim **WVERWEIS** (das W steht für waagerecht) sind die Daten in der Tabelle horizontal angeordnet; sie werden von links nach rechts durchsucht.

## Ein Beispiel für die Funktionsweise von WVERWEIS()

Ein Beispiel soll die Funktionsweise von WVERWEIS() illustrieren. In der folgenden Abbildung sollen im Bereich B2:B3 automatisch der Umsatz und Gewinn zu der jeweiligen Kalenderwoche aus der Zelle B1 übertragen werden. Die Daten (Umsatz und Gewinn) befinden sich hierbei horizontal angeordnet im Bereich E1:I3 – der Datentabelle.

[illegible]

### Anordnung der Daten für den WVERWEIS()

Um den Umsatz automatisch zu übertragen, erfassen Sie die folgende Formel in der Zelle B2 und bestätigen Sie Ihre Eingabe durch die Enter-Taste:

**=WVERWEIS(B1; \$E\$1:\$I\$3; 2; FALSCH)**

Durch die Formel wird automatisch der Umsatz aus der dritten Kalenderwoche in die Zelle B2 übertragen. Würden Sie die Kalenderwoche in der Zelle B1 ändern, dann wird automatisch der Umsatz entsprechend angepasst.

[illegible]

### Auswahl von Daten mit der Excel-Funktion WVERWEIS()

Welche Bedeutung haben die Argumente der Funktion WVERWEIS()?

- B1 = **Suchkriterium**: Ist der Wert, den Sie in der horizontalen Datentabelle nachschlagen. In dem Beispiel ist dies die Kalenderwoche 3.
- \$E\$1:\$I\$3 = **Datentabelle**: Ist der Bereich, aus dem Sie bestimmte Daten zurückgeben wollen. Sie sollten die Datentabelle immer als absoluten Bezug mit Dollarzeichen erfassen, damit sich der Bereich beim Kopieren der Funktion WVERWEIS() in weitere Zellen nicht verschiebt.
- 2 = **Zeilennummer** in der Datentabelle, die den Rückgabewert enthält. In Beispiel soll der Wert der 2. Zeile der Datentabelle (Umsatz) zurückgegeben werden.
- FALSCH = **Bereich\_Verweis**: Dieses Argument gibt an, ob eine genaue Übereinstimmung (FALSCH) oder ungefähre Übereinstimmung (WAHR) zutreffen soll. In Beispiel muss der Umsatz exakt zur Kalenderwoche stimmen, daher wurde der Wert FALSCH (genaue Übereinstimmung) als Argument gewählt.

**Merke****Gesucht wird in der ersten Zeile der Datentabelle**

Das Suchkriterium – im Beispiel die Kalenderwoche – muss immer in der ersten Zeile der Datentabelle enthalten sein! Beachten Sie außerdem, dass beim WVERWEIS() die Daten immer von links nach rechts (waagerecht) angeordnet sein müssen!

Die Funktion WVERWEIS() hat in der Datentabelle (\$E\$1:\$I\$3) in der ersten Zeile den Suchbegriff (B1 = Kalenderwoche = 3) **waagerecht** nach rechts gesucht. Das **W** in WVERWEIS() steht für waagerecht!

Findet die Funktion WVERWEIS() in der ersten Zeile der Datentabelle exakt den gesuchten Wert, dann hält sie die Suche an und wandert so viele Zeilen nach unten, wie Sie bei Zeilennummer angegeben haben.

Der Wert, der dann in der entsprechenden Zelle steht, wird als Ergebnis zurückgegeben.

D	E	F	G	H	I
<b>Kalenderwoche</b>	1	2	3	4	5
<b>Umsatz</b>	17.323	26.292	37.031	23.949	14.401
<b>Gewinn</b>	3.751	4.018	3.616	1.112	1.309

*Funktionsweise und Suchrichtung von WVERWEIS()*

Um jetzt auch noch den Gewinn in die Zelle B3 zu übertragen, müssen Sie die WVERWEIS()-Formel nur in der Zeilennummer anpassen, da der Gewinn in der 3. Zeile der Datentabelle steht.

=WVERWEIS(B1; \$E\$1:\$I\$3; **3**; FALSCH)



B3 <span>fx</span> =WVERWEIS(B1;\$E\$1:\$I\$3;3;FALSCH)									
	A	B	C	D	E	F	G	H	I
1	Kalenderwoche	3		Kalenderwoche	1	2	3	4	5
2	Umsatz	37.031		Umsatz	17.323	26.292	37.031	23.949	14.401
3	Gewinn	3.616		Gewinn	3.751	4.018	3.616	1.112	1.309
4									

Auswertung einer Datentabelle mit WVERWEIS() – Auswahl der Angabe in der dritten Zeile

## Ein Beispiel für ungefähre Übereinstimmung (WAHR)

Die ungefähre Übereinstimmung (WAHR) beim WVERWEIS() hilft Ihnen, wenn Sie Werte oder Inhalte staffelweise zuordnen wollen.

In der folgenden Abbildung sehen Sie in dem Bereich E1:I2 eine umsatzabhängige gestaffelte Bonustabelle. Um die entsprechenden Bonussätze abhängig vom Umsatz mit dem WVERWEIS() zu übernehmen, können Sie den WVERWEIS() wie in der folgenden Abbildung mit dem Argument **WAHR** verwenden.

**=WVERWEIS(B1; \$E\$1:\$I\$3; 2; WAHR)**

B2 <span>fx</span> =WVERWEIS(B1;\$E\$1:\$I\$2;2;WAHR)									
	A	B	C	D	E	F	G	H	I
1	Umsatz	120.000		Umsatz	0	50.000	100.000	150.000	200.000
2	Bonus	10,00%		Bonus	0%	5%	10%	15%	20%
3									

Auswertung einer Datentabelle mit WVERWEIS() und ungefährer Übereinstimmung beim Suchkriterium

Der WVERWEIS() geht mit dem Argument WAHR() hier wie folgt vor:

- Zuerst wird der Suchbegriff (120.000) in der 1. Zeile der Datentabelle gesucht.
- Wird der Begriff exakt gefunden, dann wird die 2. Zeile in der Spalte, in welcher der Begriff gefunden wurde, als Ergebnis zurückgegeben.
- Wird der Suchbegriff nicht exakt gefunden, dann geht WVERWEIS() alle Einträge in der 1. Zeile der Datentabelle durch, bis er auf den Wert 150.000 stößt. Dieser Wert ist größer als 120.000.
- Anschließend springt WVERWEIS() einen Eintrag zurück nach links auf die 100.000 und gibt den Bonussatz (10 %) in der 2. Zeile zurück.

**Merke****Datentabelle für das Argument WAHR nach dem Suchkriterium sortieren**

Damit Sie den WVERWEIS() mit ungefähre Übereinstimmung (WAHR) einsetzen können, müssen Sie die Datentabelle zuvor aufsteigend nach dem Suchkriterium (1. Zeile) sortieren.

## Mit der Excel-Funktion INDEX() Tabellenbereiche durchsuchen

Mit der Excel-Funktion INDEX() finden Sie in einem Tabellenbereich genau den Wert, den Sie mit der Angabe von Zeile und Spalte suchen. Wenn Sie diese Funktion mit der Funktion VERGLEICH() kombinieren, sind flexible Auswertungen von Tabellen möglich.

Zuletzt geändert am 18.03.2026



Mit den Funktionen **SVERWEIS()** und **WVERWEIS()** können Sie einen Tabellenbereich oder eine Matrix sowohl vertikal als auch horizontal nach bestimmten Kriterien durchsuchen und sich so einen bestimmten Inhalt aus dem Tabellenbereich zurückgeben lassen.

Hierbei funktioniert die Suche aber immer nur in eine Richtung: horizontal oder vertikal. Je nachdem, welche von den beiden Funktionen Sie gerade einsetzen.

Manchmal kommt es jedoch vor, dass Sie einen Tabellenbereich sowohl horizontal als auch vertikal nach einem bestimmten Wert durchsuchen müssen. Hierfür können Sie die Funktion **INDEX()** in Excel verwenden.

### Ein einfaches Beispiel, um die Funktionsweise von INDEX() zu erklären

In der folgenden Abbildung sehen Sie eine Matrix, in der Umsatzwerte von Geschäftsfilialen pro Kalendermonat dargestellt sind. Die Monate sind hierbei horizontal (S2 bis S13), während die Filialen vertikal angeordnet sind (Z2 bis Z13).

Aus Vereinfachungsgründen werden als Zeilen- und Spaltenbeschriftung (gelbe Markierung) lediglich die jeweilige Zeilen- (Z2 = Zeile 2) und Spaltennummer (S2 = Spalte 2) verwendet.

Es soll nun im ersten Beispiel der Wert der grün markierten Zelle mit der Formel **INDEX()** ausgelesen werden. Er befindet sich in der Schnitzelle der 4. Zeile und 3. Spalte (Z4 – S3).

C18													
	A	B	C	D	E	F	G	H	I	J	K	L	M
1		S2	S3	S4	S5	S6	S7	S8	S9	S10	S11	S12	S13
2	Z2	837	646	427	894	832	767	322	180	169	864	297	513
3	Z3	477	361	186	822	852	533	259	158	744	159	870	611
4	Z4	313	125	911	457	992	689	420	470	286	764	962	526
5	Z5	565	344	762	258	129	480	682	575	941	445	242	881
6	Z6	312	933	306	453	683	593	203	906	219	281	742	639
7	Z7	130	829	396	212	240	944	735	387	476	583	559	268
8	Z8	574	540	714	413	187	298	183	320	638	383	885	871
9	Z9	699	910	328	360	840	855	635	101	438	491	948	520
10	Z10	323	717	430	235	264	781	140	662	102	798	612	950
11	Z11	257	765	524	558	597	539	780	181	814	190	605	592
12	Z12	602	191	836	980	154	99	175	915	464	665	636	411
13	Z13	807	267	245	799	516	577	588	797	103	309	853	465
14													
15		Zeile	4										
16		Spalte	3										
17													
18		Ergebnis											
19													

*Tabellenbereich oder Matrix als Suchbereich für INDEX()*

Um die Formel bezüglich der Koordinaten der jeweiligen Werte dynamisch zu gestalten, sollen die jeweiligen Zeilen- und Spaltenwerte in den Zellen C15 (Zeilennummer) und C16 (Spaltennummer) erfasst werden. Dort tragen Sie also ein, in welcher Zeile und welcher Spalte sich der gesuchte Wert befindet.

Den Monatsumsatz der gewünschten Filiale (in der grünen Zelle) erhalten Sie dann mit der folgenden Formel, die Sie in die Zelle C18 eingeben und durch die Enter-Taste bestätigen:

**=INDEX(A1:M13; C15; C16)**

Durch die Formel wird automatisch der Umsatz aus der 4. Zeile und der 3. Spalte in die Zelle C18 übertragen. Würden Sie die Zeilen- und Spaltennummern in den Zellen C15 und C16 ändern, dann wird automatisch der Umsatz entsprechend angepasst.

C18													
=INDEX(A1:M13;C15;C16)													
	A	B	C	D	E	F	G	H	I	J	K	L	M
1		S2	S3	S4	S5	S6	S7	S8	S9	S10	S11	S12	S13
2	Z2	837	646	427	894	832	767	322	180	169	864	297	513
3	Z3	477	361	186	822	852	533	259	158	744	159	870	611
4	Z4	313	125	911	457	992	689	420	470	286	764	962	526
5	Z5	565	344	762	258	129	480	682	575	941	445	242	881
6	Z6	312	933	306	453	683	593	203	906	219	281	742	639
7	Z7	130	829	396	212	240	944	735	387	476	583	559	268
8	Z8	574	540	714	413	187	298	183	320	638	383	885	871
9	Z9	699	910	328	360	840	855	635	101	438	491	948	520
10	Z10	323	717	430	235	264	781	140	662	102	798	612	950
11	Z11	257	765	524	558	597	539	780	181	814	190	605	592
12	Z12	602	191	836	980	154	99	175	915	464	665	636	411
13	Z13	807	267	245	799	516	577	588	797	103	309	853	465
14													
15		Zeile	4										
16		Spalte	3										
17													
18		Ergebnis	125										
19													

Mit INDEX() einen Matrixwert ermitteln

## Wie ist die Funktion INDEX() aufgebaut?

Schauen wir uns die Funktionsweise etwas genauer an, indem wir die Argumente der Funktion **INDEX()** betrachten:

- A1:M13 = **Matrix**: Ist die Datentabelle, aus der Sie bestimmte Daten zurückgeben wollen.
- C15 = **Zeile**: Zeilennummer der Matrix, die den gesuchten Wert enthält.
- C16 = **Spalte**: Spaltennummer der Matrix, die den gesuchten Wert enthält.

Die Funktion INDEX() hat nun in der Tabelle A1:M13 (Matrix) den Wert der Schnitzelle ausgelesen, in der sich die angegebene Zeilen- und Spaltennummer schneiden.

**Wichtig:** Die Zählung der Zeile und Spalte beginnt mit der ersten Zelle des Bereichs, der bei Matrix angegeben ist. Im Beispiel also bei Zelle A1. Das gilt auch dann, wenn in der ersten Zeile und Spalte die jeweiligen Überschriften stehen; im Beispiel Z1 bis Z13 und S1 bis S13 (gelbe Zellen).

## INDEX() mit Bezeichnungen anstelle von Nummern einsetzen

Mit der **Zeilen- und Spaltennummer** können Sie mit der Funktion INDEX() aus einer Matrix einen bestimmten Wert herauslesen. Oft ist diese Vorgehensweise aber nicht praktikabel, da nicht bekannt ist, an welcher Position (Zeile oder Spalte) innerhalb eines Bereichs sich der gesuchte Wert befindet.

Einfacher wäre es, wenn analog zu SVERWEIS() und WVERWEIS() auch **Bezeichnungen anstelle von Zeilen- und Spaltennummern** verwendet werden können.

Dazu wandeln wir das Ausgangsbeispiel ab, indem wir die Spaltennummer durch den jeweiligen Monatsnamen als Kürzel ersetzen. Die Auswahl soll jetzt auch durch die Eingabe des Monatsnamens und nicht mehr durch die Spaltennummer erfolgen.

Im folgenden Beispiel soll der Umsatz des Monats November aus der 10. Zeile ausgelesen werden.

	A	B	C	D	E	F	G	H	I	J	K	L	M
1		Jan	Feb	Mrz	Apr	Mai	Jun	Jul	Aug	Sep	Okt	Nov	Dez
2	Z2	837	646	427	894	832	767	322	180	169	864	297	513
3	Z3	477	361	186	822	852	533	259	158	744	159	870	611
4	Z4	313	125	911	457	992	689	420	470	286	764	962	526
5	Z5	565	344	762	258	129	480	682	575	941	445	242	881
6	Z6	312	933	306	453	683	593	203	906	219	281	742	639
7	Z7	130	829	396	212	240	944	735	387	476	583	559	268
8	Z8	574	540	714	413	187	298	183	320	638	383	885	871
9	Z9	699	910	328	360	840	855	635	101	438	491	948	520
10	Z10	323	717	430	235	264	781	140	662	102	798	612	950
11	Z11	257	765	524	558	597	539	780	181	814	190	605	592
12	Z12	602	191	836	980	154	99	175	915	464	665	636	411
13	Z13	807	267	245	799	516	577	588	797	103	309	853	465
14													
15		Zeile	10										
16		Spalte	Nov										
17													
18	Ergebnis												
19													

Suche in Matrix mit Bezeichnungen für die Spalten statt Spaltennummern

Da die Funktion INDEX() als Argument die Spaltennummer benötigt, braucht es eine Lösung, mit der sich die Spaltennummer aus dem Monatsnamen ergibt. Hier kommt die Funktion **VERGLEICH()** ins Spiel.

Mit der Funktion **VERGLEICH()** können Sie in einem Bereich nach einem angegebenen Element (zum Beispiel Monatsname) suchen und anschließend die relative Position als Zahl dieses Elements im Bereich zurückgegeben.

Ersetzen Sie daher das Spaltenargument C16 in der Formel =INDEX(A1:M13; C15; **C16**) durch die folgende Funktion:

**=VERGLEICH(C16; A1:M1; 0)**

## Was macht die Excel-Funktion VERGLEICH()?

Die Funktion VERGLEICH() basiert auf folgenden Argumenten:

- C16 = **Suchkriterium**: Ist der Wert, den Sie in der Suchmatrix suchen.
- A1:M1 = **Suchmatrix**: Ist der Tabellenbereich, in dem das Suchkriterium gesucht wird; im Beispiel ist das der Bereich für die Spaltenüberschrift mit den abgekürzten Monatsnamen (Jan etc.).
- 0 = **Vergleichstyp**: Der Vergleichstyp 0 bedeutet, dass der gesuchte Wert exakt in der Suchmatrix erscheinen muss.

Die Funktion **=VERGLEICH(C16; A1:M1; 0)** sucht demnach im Bereich A1:M1 nach dem Begriff **Nov** (C16) und gibt die relative Position des Suchbegriffs als Zahl aus. Der Begriff **Nov** steht im Bereich A1:M1 an 12. Stelle. Es wird somit die Zahl 12 als Ergebnis zurückgegeben.

Das ist der Wert der Spalte, der für die INDEX()-Formel gebraucht wird. Integrieren Sie diese VERGLEICH()-Formel anstelle des Bezugs C16 in die INDEX()-Formel. Sie können somit direkt nach Monatsnamen anstelle von Spaltennummern die Matrix durchsuchen lassen.

C18														=INDEX(A1:M13;C15;VERGLEICH(C16;A1:M1;0))	
	A	B	C	D	E	F	G	H	I	J	K	L	M		
1		Jan	Feb	Mrz	Apr	Mai	Jun	Jul	Aug	Sep	Okt	Nov	Dez		
2	Z2	837	646	427	894	832	767	322	180	169	864	297	513		
3	Z3	477	361	186	822	852	533	259	158	744	159	870	611		
4	Z4	313	125	911	457	992	689	420	470	286	764	962	526		
5	Z5	565	344	762	258	129	480	682	575	941	445	242	881		
6	Z6	312	933	306	453	683	593	203	906	219	281	742	639		
7	Z7	130	829	396	212	240	944	735	387	476	583	559	268		
8	Z8	574	540	714	413	187	298	183	320	638	383	885	871		
9	Z9	699	910	328	360	840	855	635	101	438	491	948	520		
10	Z10	323	717	430	235	264	781	140	662	102	798	612	950		
11	Z11	257	765	524	558	597	539	780	181	814	190	605	592		
12	Z12	602	191	836	980	154	99	175	915	464	665	636	411		
13	Z13	807	267	245	799	516	577	588	797	103	309	853	465		
14															
15		Zeile	10												
16		Spalte	Nov												
17															
18		Ergebnis	612												
19															

Mit INDEX() und VERGLEICH() einen Matrixwert mit Spaltenbezeichnungen ermitteln

Mit der gleichen Vorgehensweise können Sie auch die Zeilennummer in der INDEX()-Formel ersetzen. Ersetzen Sie das Zeilenargument (C15) durch die folgende VERGLEICH()-Funktion:

**=VERGLEICH(C15; A1:A13; 0)**

Die **modifizierte INDEX()-Formel** sieht dann wie folgt aus:

**=INDEX(A1:M13; VERGLEICH(C15; A1:A13; 0); VERGLEICH(C16; A1:M1; 0))**

Durch die Integration der VERGLEICH()-Funktionen sind Sie somit komplett unabhängig von den jeweiligen Zeilen- und Spaltennummern.



C18		=INDEX(A1:M13;VERGLEICH(C15;A1:A13;0);VERGLEICH(C16;A1:M1;0))													
	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O
1		Jan	Feb	Mrz	Apr	Mai	Jun	Jul	Aug	Sep	Okt	Nov	Dez		
2	Nürnberg	837	646	427	894	832	767	322	180	169	864	297	513		
3	München	477	361	186	822	852	533	259	158	744	159	870	611		
4	Würzburg	313	125	911	457	992	689	420	470	286	764	962	526		
5	Hamburg	565	344	762	258	129	480	682	575	941	445	242	881		
6	Berlin	312	933	306	453	683	593	203	906	219	281	742	639		
7	Köln	130	829	396	212	240	944	735	387	476	583	559	268		
8	Dortmund	574	540	714	413	187	298	183	320	638	383	885	871		
9	Bochum	699	910	328	360	840	855	635	101	438	491	948	520		
10	Bielefeld	323	717	430	235	264	781	140	662	102	798	612	950		
11	Freiburg	257	765	524	558	597	539	780	181	814	190	605	592		
12	Augsburg	602	191	836	980	154	99	175	915	464	665	636	411		
13	Bremen	807	267	245	799	516	577	588	797	103	309	853	465		
14															
15		Zeile	Köln												
16		Spalte	Jun												
17															
18		Ergebnis	944												
19															

Tabellenbereiche flexibel mit den Funktionen INDEX() und VERGLEICH() durchsuchen

# Wert in einer Excel-Tabelle suchen und Zelle anzeigen lassen

Sie suchen einen bestimmten Wert, Text, Betrag oder Zahl in einer Liste (Spalte) und wollen wissen, in welcher Zelle er auftaucht? Mit dieser Formel ermitteln Sie die Zelladresse.

Zuletzt geändert am 18.03.2026



In Excel geht es oft nicht nur darum, Inhalte zu finden, sondern auch darum, deren **genaue Position** zu ermitteln. Sie wollen wissen, wo ein bestimmter Wert in einer vorgegebenen Tabelle oder Liste zu finden ist.

Excel bietet dafür verschiedene Funktionen, mit denen Sie die Adresse einer Zelle als Suchergebnis zurückgeben können. In diesem Beitrag erfahren Sie, wie Sie mit einer einfachen Formel die Zelladresse eines Suchergebnisses in Excel ermitteln.

## Zelladresse ermitteln

Mit dieser Excel-Formel ermitteln Sie die **Zelladresse**, in der sich der von Ihnen gesuchte Wert befindet:

**=ZELLE("adresse"; XVERWEIS(C2; A2:A13; A2:A13))**

Die Zellbezüge in der Formel haben dabei die folgende Bedeutung:

- **"adresse"**: Parameter der Funktion ZELLE() der besagt, dass die Information „Zelladresse“ ausgegeben werden soll
- C2 = **Suchkriterium**: Wert (Zahl oder Text), dessen Zelladresse gesucht werden soll
- A2:A13 = **Suchspalte**: In dieser Spalte wird der gesuchte Wert gesucht
- A2:A13 = **Ergebnisspalte**: Aus dieser Spalte wird der Wert zurückgegeben, der dem entsprechenden Wert in der Suchspalte entspricht

Suchspalte und Ergebnisspalte sind identisch, da die Zelladresse von einem bestimmten Wert ermittelt werden soll.

C5               =ZELLE("adresse";XVERWEIS(C2;A2:A13;A2:A13))						
	A	B	C	D	E	
1	<b>Wert</b>		<b>gesuchter Wert</b>			
2	1465		1410			
3	1732					
4	1142		<b>Zelladresse</b>			
5	1596		\$A\$10			
6	1738					
7	1632					
8	1331					
9	1857					
10	1410					
11	1498					
12	1442					
13	1850					

*Formel zur Bestimmung der Zelladresse eines gesuchten Werts*

Die Funktion XVERWEIS(C2; A2:A13; A2:A13) ermittelt zunächst die Position des gesuchten Werts und übergibt diese Information an die Funktion ZELLE().

Die Funktion ZELLE() gibt daraufhin die entsprechende Zellreferenz als Ergebnis zurück: =**ZELLE("adresse"; XVERWEIS(C2; A2:A13; A2:A13))**.

**Hinweis:** Taucht der gesuchte Wert in der Liste mehrmals auf, wird nur die erste Fundstelle von oben ausgegeben. Sie können nur in einer einspaltigen Liste, nicht in einem mehrspaltigen Bereich suchen.

# Suchen nach ungefährer Übereinstimmung in Excel-Listen

So können Sie in einer unsortierten Liste die Übereinstimmung finden, die am nächsten an einem gesuchten Wert dran ist. Eine Formel mit XVERWEIS() und ABS() liefert den gesuchten Eintrag in einer Liste.

Zuletzt geändert am 18.03.2026



Die Funktion **SVERWEIS()** bietet die Option, **Listen mit ungefährer Übereinstimmung** zu durchsuchen. Dies setzt aber voraus, dass die erste Spalte in der Tabelle entweder numerisch oder alphabetisch sortiert ist. Nur dann sucht SVERWEIS() nach dem am nächsten liegenden Wert.

Diese Option ist sehr nützlich, kann aber nicht mit den folgenden Eigenschaften einer Datentabelle umgehen:

- Liste oder Datentabelle ist nicht sortiert.
- Es gibt Abweichungen mit unterschiedlichen Vorzeichen (für die Auswertung ist der absolute Wert maßgeblich)

## Beispiel für unsortierte Listen zur „ungefähren Auswertung“

In der folgenden Abbildung sehen Sie eine Liste von Städten mit den jeweiligen Einwohnerzahlen. Die Liste ist nicht sortiert.

Die Aufgabe besteht darin, die nächste Übereinstimmung zu einem in der Zelle D1 eingegebenen Zielwert (Einwohnerzahl) zu finden; also die Stadt auszuwählen, die „**in etwa**“ die angegebene Einwohnerzahl hat.

	A	B	
1	<b>Stadt</b>	<b>Einwohnerzahl (ca.)</b>	
2	Tokio	37.500.000	
3	New York City	8.400.000	
4	Berlin	3.700.000	
5	Sydney	5.300.000	
6	Kapstadt	4.800.000	
7	São Paulo	12.400.000	
8	Peking	21.500.000	
9	Mumbai	20.700.000	
10	Paris	11.100.000	
11	Istanbul	15.500.000	
12			
13			

Liste mit Daten für die Auswertung

## Ungefähre Auswahl mit der Funktion XVERWEIS()

Obwohl es vielleicht nicht so aussieht, ist dies im Wesentlichen ein Nachschlageproblem. Hierfür können Sie die Funktion XVERWEIS() wie folgt einsetzen:

**=XVERWEIS(0; ABS(B2:B11-D1); A2:A11; ; 1)**

Die Formel geht von den folgenden Annahmen aus:

- **B2:B11**: In diesen Zellen befinden sich die Einwohnerzahlen.
- **A2:A11**: Stadtnamen, die zurückgegeben werden sollen.
- **D1**: Wert, mit dem die Einträge (Einwohnerzahlen) verglichen werden und von dem das Ergebnis möglichst wenig abweichen soll.

E1		✕ ✓ <i>fx</i>		=XVERWEIS(0;ABS(B2:B11-D1);A2:A11;;1)			
	A	B	C	D	E	F	G
1	<b>Stadt</b>	<b>Einwohnerzahl (ca.)</b>		5.000.000	Kapstadt		
2	Tokio	37.500.000					
3	New York City	8.400.000					
4	Berlin	3.700.000					
5	Sydney	5.300.000					
6	Kapstadt	4.800.000					
7	São Paulo	12.400.000					
8	Peking	21.500.000					
9	Mumbai	20.700.000					
10	Paris	11.100.000					
11	Istanbul	15.500.000					
12							

Auswertung der Liste mit XVERWEIS()

## Wie die Formel genau funktioniert

Zuerst wird die Differenz von jeder einzelnen Zelle im Bereich B2:B11 zur gesuchten Einwohnerzahl in der Zelle D1 ermittelt. Diese Differenz kann sowohl eine positive als auch eine negative Zahl sein.

=XVERWEIS(0; ABS(**B2:B11-D1**); A2:A11; ; 1)

Die einzelnen Differenzen werden anschließend an die Funktion **ABS()** weitergeleitet. Die Funktion ABS() in Excel gibt den absoluten Wert einer Zahl zurück. Das bedeutet, sie entfernt das Vorzeichen einer Zahl, sodass immer ein positiver Wert ausgegeben wird.

Sie haben somit eine Liste von positiven Differenzwerten, welche die Suchmatrix in der Funktion XVERWEIS() darstellt.

=XVERWEIS(0; **ABS(B2:B11-D1)**; A2:A11; ; 1)

In dieser Liste (positive Differenzen) wird jetzt nach dem Suchkriterium gesucht. Das Suchkriterium ist der Wert 0 im ersten Argument der Funktion **XVERWEIS()**, denn die Differenz zur gesuchten Einwohnerzahl (5.000.000) soll möglichst gering oder eben 0 sein.

Bei der Suche wird nun der **Vergleichsmodus 1** angewendet. Der Vergleichsmodus 1 sucht nach einer genauen Übereinstimmung. Wird keine genaue Übereinstimmung gefunden, dann wird das nächstgrößere Element zurückgegeben.

=XVERWEIS(; ABS(B2:B11-D1); A2:A11; ; **1**)

Zum Schluss wird dann der entsprechende Wert zurückgegeben, der sich in der Rückgabematrix (A2:A11) der Funktion XVERWEIS() an der entsprechenden Stelle befindet.

Im Beispiel steht der Wert mit der kleinsten Abweichung an der fünften Stelle in der Suchmatrix (B2:B11). Daher wird der fünfte Wert in der Rückgabeliste A2:A11 als Ergebnis zurückgegeben. Dies ist im Beispiel der Wert **Kapstadt**.

=XVERWEIS(0; ABS(B2:B11-D1); **A2:A11**; ; 1)

## Mit der Excel-Funktion EINDEUTIG() Listen auswerten

Welche Einträge tauchen in einer Liste oder Matrix genau einmal oder mindestens einmal auf? Das lässt sich mit der Excel-Funktion EINDEUTIG() einfach und schnell ermitteln.

Zuletzt geändert am 18.03.2026



Eine typische Excel-Aufgabe aus der Praxis: Sie sollen aus einer Liste die eindeutigen Einträge ermitteln.

**Beispiel:** Sie haben eine Auflistung der verkauften Produkte des heutigen Tages und sollen der Verkaufsleitung mitteilen, welche unterschiedlichen Produkte heute verkauft wurden.

Je nach Umfang der Liste kann die Aufgabe zeitaufwendig und mühsam werden. Doch mit der Excel-Funktion EINDEUTIG() wird diese Aufgabe zum Kinderspiel.

### Was die Excel-Funktion EINDEUTIG() leistet

Mit der Funktion **EINDEUTIG()** können Sie eine Liste oder Tabelle auf eindeutige Einträge überprüfen und diese dann an einer anderen Stelle ausgeben. Es werden also mit der Funktion EINDEUTIG() alle doppelten Einträge entfernt und nur die Unikate (eindeutige Einträge) als Ergebnis ausgegeben.

EINDEUTIG() ist eine dynamische Funktion, deren Ergebnis sich automatisch anpasst, sobald Änderungen an der Ausgangsliste vorgenommen werden.

Das folgende Beispiel zeigt die Funktionsweise von EINDEUTIG().



	A	B	C	D	E	F
1	Produkt A		Produkt A			
2	Produkt A		Produkt B			
3	Produkt B		Produkt E			
4	Produkt A		Produkt Z			
5	Produkt A					
6	Produkt A					
7	Produkt A					
8	Produkt B					
9	Produkt A					
10	Produkt A					
11	Produkt B					
12	Produkt A					
13	Produkt A					
14	Produkt A					
15	Produkt A					
16	Produkt E					
17	Produkt A					
18	Produkt A					
19	Produkt A					
20	Produkt A					
21	Produkt Z					

Beispiel für Excel-Funktion EINDEUTIG()

## Die Parameter der Funktion EINDEUTIG()

Sie sehen am linken Rand der vorigen Abbildung die Verkaufsliste, aus der nun ermittelt werden soll, welche Produkte verkauft wurden. Hierfür wird in der Zelle C1 die Funktion eingegeben:

**=EINDEUTIG(A1:A21; FALSCH; FALSCH)**

Bei **EINDEUTIG()** handelt es sich um eine **Array-Funktion**, die bei mehreren Ergebnissen die jeweiligen Treffer in den entsprechenden Zellen darunter auflistet. Ist nicht genügend Platz, gibt es eine Fehlermeldung: #ÜBERLAUF!

Schauen wir uns als Nächstes die einzelnen Argumente von EINDEUTIG() genauer an:

- **Liste oder Matrix:** Der Bereich, der auf eindeutige Einträge hin untersucht werden soll. Dies kann ein Zellbereich oder eine einzelne Zeile oder Spalte sein. Im Beispiel der Bereich A1:A21.
- **FALSCH** (optional): Gibt an, ob die Prüfung zeilenweise (FALSCH) oder spaltenweise (WAHR) durchgeführt werden soll. Wird dieses Argument nicht angegeben, dann wird standardmäßig zeilenweise (FALSCH) geprüft.
- **FALSCH** (optional): Es wird jeder Eintrag aufgelistet, der in der Ursprungsliste **mindestens einmal** vorkommt. Wird dieses Argument nicht angegeben, dann wird standardmäßig jeder Wert einmal aufgelistet (FALSCH). Bei **WAHR** wird ein Eintrag ausgegeben, wenn er **nur genau einmal** in der Liste vorkommt.

## Auswertung nach Spalten einer Matrix

Welche Effekte es hat, wenn Sie eine Matrix nach Spalten mit EINDEUTIG() auswerten, zeigt das folgende Beispiel:

- Die Matrix im Bereich A1:D21 enthält vier Spalten; die Spalten A und B sind identisch.
- Dann werten Sie diese Matrix aus mit der Funktion in Zelle F1:  
=EINDEUTIG(A1:D21; **WAHR**; FALSCH)
- Tragen Sie als zweiten Wahrheitswert WAHR ein, prüft EINDEUTIG() die gesamten Spalten und vergleicht diese miteinander.
- Da die Spalten A und B identisch sind, wird sie nur einmal aufgelistet; im Bereich F1:F21.
- Die Spalten C und D unterscheiden sich; sie werden deshalb beide vollständig in den Spalten G und H aufgelistet.

Das Ergebnis zeigt die folgende Abbildung.

	A	B	C	D	E	F	G	H
1	Produkt A	Produkt A	Produkt C	Produkt A		Produkt A	Produkt C	Produkt A
2	Produkt A	Produkt A	Produkt C	Produkt A		Produkt A	Produkt C	Produkt A
3	Produkt B	Produkt B	Produkt B	Produkt B		Produkt B	Produkt B	Produkt B
4	Produkt A	Produkt A	Produkt C	Produkt A		Produkt A	Produkt C	Produkt A
5	Produkt A	Produkt A	Produkt C	Produkt A		Produkt A	Produkt C	Produkt A
6	Produkt A	Produkt A	Produkt C	Produkt A		Produkt A	Produkt C	Produkt A
7	Produkt A	Produkt A	Produkt C	Produkt A		Produkt A	Produkt C	Produkt A
8	Produkt B	Produkt B	Produkt B	Produkt B		Produkt B	Produkt B	Produkt B
9	Produkt A	Produkt A	Produkt C	Produkt A		Produkt A	Produkt C	Produkt A
10	Produkt A	Produkt A	Produkt C	Produkt A		Produkt A	Produkt C	Produkt A
11	Produkt B	Produkt B	Produkt B	Produkt B		Produkt B	Produkt B	Produkt B
12	Produkt A	Produkt A	Produkt C	Produkt A		Produkt A	Produkt C	Produkt A
13	Produkt A	Produkt A	Produkt C	Produkt D		Produkt A	Produkt C	Produkt D
14	Produkt A	Produkt A	Produkt C	Produkt D		Produkt A	Produkt C	Produkt D
15	Produkt A	Produkt A	Produkt C	Produkt D		Produkt A	Produkt C	Produkt D
16	Produkt E	Produkt E	Produkt E	Produkt E		Produkt E	Produkt E	Produkt E
17	Produkt A	Produkt A	Produkt C	Produkt D		Produkt A	Produkt C	Produkt D
18	Produkt A	Produkt A	Produkt C	Produkt D		Produkt A	Produkt C	Produkt D
19	Produkt A	Produkt A	Produkt C	Produkt D		Produkt A	Produkt C	Produkt D
20	Produkt A	Produkt A	Produkt C	Produkt D		Produkt A	Produkt C	Produkt D
21	Produkt Z	Produkt Z	Produkt Z	Produkt Z		Produkt Z	Produkt Z	Produkt Z
22								

Excel-Funktion EINDEUTIG() mit Auswertung anhand Spalte

## Auswertung nach Zeilen einer Matrix

Im folgenden Beispiel wird die Matrix nach Zeilen ausgewertet.

- Die Matrix im Bereich A1:D21 enthält 21 Zeilen; unter anderem sind Zeile 1 und Zeile 2 identisch.
- Dann werten Sie diese Liste aus mit der Funktion in Zelle J1:  
=EINDEUTIG(A1:D21; **FALSCH**; FALSCH)
- Tragen Sie als zweiten Wahrheitswert FALSCH ein, prüft EINDEUTIG() die gesamten Zeilen und vergleicht diese miteinander.
- Da unter anderem die Zeilen 1 und 2 identisch sind, wird dieses Ergebnis nur einmal aufgelistet; im Bereich J1:M1.
- Entsprechend gibt es in der Matrix noch vier weitere unterschiedliche Zeilen, die ab Zelle J2 aufgelistet werden.

Excel-Screenshot: Formelband zeigt die Funktion `=EINDEUTIG(A1:D21;FALSCH;FALSCH)`. Die Tabelle enthält Daten für Produkte A bis Z.

	A	B	C	D	E	F	G	H	I	J	K	L	M
1	Produkt A	Produkt A	Produkt C	Produkt A		Produkt A	Produkt C	Produkt A		Produkt A	Produkt A	Produkt C	Produkt A
2	Produkt A	Produkt A	Produkt C	Produkt A		Produkt A	Produkt C	Produkt A		Produkt B	Produkt B	Produkt B	Produkt B
3	Produkt B	Produkt B	Produkt B	Produkt B		Produkt B	Produkt B	Produkt B		Produkt A	Produkt A	Produkt C	Produkt D
4	Produkt A	Produkt A	Produkt C	Produkt A		Produkt A	Produkt C	Produkt A		Produkt E	Produkt E	Produkt E	Produkt E
5	Produkt A	Produkt A	Produkt C	Produkt A		Produkt A	Produkt C	Produkt A		Produkt Z	Produkt Z	Produkt Z	Produkt Z
6	Produkt A	Produkt A	Produkt C	Produkt A		Produkt A	Produkt C	Produkt A					
7	Produkt A	Produkt A	Produkt C	Produkt A		Produkt A	Produkt C	Produkt A					
8	Produkt B	Produkt B	Produkt B	Produkt B		Produkt B	Produkt B	Produkt B					
9	Produkt A	Produkt A	Produkt C	Produkt A		Produkt A	Produkt C	Produkt A					
10	Produkt A	Produkt A	Produkt C	Produkt A		Produkt A	Produkt C	Produkt A					
11	Produkt B	Produkt B	Produkt B	Produkt B		Produkt B	Produkt B	Produkt B					
12	Produkt A	Produkt A	Produkt C	Produkt A		Produkt A	Produkt C	Produkt A					
13	Produkt A	Produkt A	Produkt C	Produkt D		Produkt A	Produkt C	Produkt D					
14	Produkt A	Produkt A	Produkt C	Produkt D		Produkt A	Produkt C	Produkt D					
15	Produkt A	Produkt A	Produkt C	Produkt D		Produkt A	Produkt C	Produkt D					
16	Produkt E	Produkt E	Produkt E	Produkt E		Produkt E	Produkt E	Produkt E					
17	Produkt A	Produkt A	Produkt C	Produkt D		Produkt A	Produkt C	Produkt D					
18	Produkt A	Produkt A	Produkt C	Produkt D		Produkt A	Produkt C	Produkt D					
19	Produkt A	Produkt A	Produkt C	Produkt D		Produkt A	Produkt C	Produkt D					
20	Produkt A	Produkt A	Produkt C	Produkt D		Produkt A	Produkt C	Produkt D					
21	Produkt Z	Produkt Z	Produkt Z	Produkt Z		Produkt Z	Produkt Z	Produkt Z					

Excel-Funktion EINDEUTIG() mit Auswertung anhand Zeile

Mit solchen zeilen- oder spaltenweisen Auswertungen einer Matrix können Sie beispielsweise eine Warenkorbanalyse durchführen. Wenn Kunden mehrere Produkte kaufen, ermitteln Sie, welche Produktkombinationen in einem Warenkorb landen.

## Mit der Excel-Funktion ERSTERWERT() Listen einfach auswerten

Mit ERSTERWERT() können Sie komplexe WENN-Funktionen vermeiden. Statt mehrfach verschachtelter Wenn-Dann-Abfragen reihen Sie die Abfragekriterien mit ERSTERWERT() einfach hintereinander an. Ein Beispiel zeigt, wie einfach das ist.

Zuletzt geändert am 18.03.2026



**ERSTERWERT()** ist eine für die tägliche Praxis nützliche Excel-Funktion, weil sich durch sie verschachtelte und komplexe WENN-Funktionen vermeiden lassen.

Damit können Sie einen **Ausgangswert mit einer Reihe von anderen Werten vergleichen**. Die Funktion gibt dann ein Ergebnis zurück, sobald die erste Übereinstimmung auftritt. Anschließend wird nicht weiter verglichen.

Außerdem können Sie mit ERSTERWERT() einen optionalen Standardwert zurückgeben, falls es keine Übereinstimmung geben sollte. ERSTERWERT() besitzt somit standardmäßig eine „eingebaute“ Fehlerroutine.

### Die Parameter der Funktion ERSTERWERT()

Die allgemeine Funktion ERSTERWERT() sieht folgendermaßen aus:

**=ERSTERWERT(Ausdruck; Wert für Übereinstimmung1...[2-126]; Rückgabewert bei Übereinstimmung1...[2-126]; Rückgabewert bei fehlender Übereinstimmung)**

Dabei bedeuten die Parameter im Einzelnen:

- **Ausdruck:** Der auszuwertende Zellinhalt.
- **Wert für Übereinstimmung1:** Der erste mit dem Ausdruck zu vergleichende Wert. Die Funktion prüft, wo der Ausdruck eingetragen ist, stoppt an der entsprechenden Stelle und gibt den folgenden Rückgabewert aus.

- **Rückgabewert bei Übereinstimmung1:** Der Wert, der ausgegeben werden soll, wenn der Wert für Übereinstimmung1 mit dem Wert von Ausdruck übereinstimmt.
- **Wert für Übereinstimmung[2-126]:** Weitere Werte, die mit dem Ausdruck verglichen werden sollen.
- **Rückgabewert bei Übereinstimmung[2-126]:** Der Wert, der ausgegeben werden soll, wenn der Wert für Übereinstimmung[2-126] mit dem Wert von Ausdruck übereinstimmt.
- **Rückgabewert bei fehlender Übereinstimmung:** Der Wert, der ausgegeben werden soll, falls es keine Übereinstimmung geben sollte.

Zwischen dem ersten Parameter (Ausdruck) und dem letzten Parameter (Rückgabewert bei fehlender Übereinstimmung) werden alle Vergleiche immer **als Paare eingegeben**:

- zunächst der zu vergleichende Wert,
- dann der Rückgabewert, der angezeigt werden soll.

Alle Parameter werden durch Semikolon ; getrennt. Mit ERSTERWERT() können Sie **bis zu 126 übereinstimmende Werte und Ergebnisse auswerten!**

## Ein Beispiel für die Excel-Funktion ERSTERWERT()

Das folgende Beispiel zeigt die Funktionsweise und Vorteile von ERSTERWERT(). In der folgenden Abbildung sehen Sie die besten 10 Teilnehmer vom Ironman-Wettbewerb auf Hawaii im Jahr 2022. Die Teilnehmer sind nach Namen sortiert.

In der Spalte C (Medaille) soll jetzt automatisch mit einer Formel – abhängig vom Rang in der Spalte A – die jeweilige Medaille ausgegeben werden: Gold für Rang 1, Silber für Rang 2, Bronze für Rang 3. Für die Teilnehmer ab Rang 4 soll eine leere Zelle ausgegeben werden.

	A	B	C	D
1	Rang	Name	Medaille	
2	9	Clement Mignon		
3	1	Gustav Iden		
4	5	Joe Skipper		
5	3	Kristian Blummenfelt		
6	7	Leon Chevalier		
7	8	Magnus Ditlev		
8	4	Max Neumann		
9	10	Patrick Lange		
10	2	Sam Laidlow		
11	6	Sebastian Kienle		
12				
13				

Liste zur Auswertung mit ERSTERWERT()

Wenn Sie diese Aufgabenstellung mit einer „klassischen“ WENN-Funktion lösen wollen, dann müssten Sie hierfür die folgende, verschachtelte WENN-Formel in der Zelle C2 erfassen:

**=WENN(A2=1; "Gold"; WENN(A2=2; "Silber"; WENN(A2=3; "Bronze"; "")))**

Und in den folgenden Zellen C3 bis C11 entsprechend.

C2		fx		=WENN(A2=1;"Gold";WENN(A2=2;"Silber";WENN(A2=3;"Bronze";"")))				
	A	B	C	D	E	F	G	H
1	Rang	Name	Medaille					
2	9	Clement Mignon						
3	1	Gustav Iden	Gold					
4	5	Joe Skipper						
5	3	Kristian Blummenfelt	Bronze					
6	7	Leon Chevalier						
7	8	Magnus Ditlev						
8	4	Max Neumann						
9	10	Patrick Lange						
10	2	Sam Laidlow	Silber					
11	6	Sebastian Kienle						
12								

Auswertung der Liste mit der Funktion WENN()

Bei einer verschachtelten WENN-Funktion verliert man schnell den Überblick: Welche Bedingungen gehören zu welcher WENN-Abfrage? Sind alle Klammern richtig gesetzt?

Deutlich einfacher ist es, wenn Sie für die Aufgabe aus dem Beispiel die Funktion ERSTERWERT() einsetzen. Für die Zuweisung der Medaillen lautet die Auswertung mit der Funktion **ERSTERWERT()** wie folgt:

**=ERSTERWERT(A2; 1; "Gold"; 2; "Silber"; 3; "Bronze"; "")**

C2		fx		=ERSTERWERT(A2;1;"Gold";2;"Silber";3;"Bronze";"")			
	A	B	C	D	E	F	G
1	Rang	Name	Medaille				
2	9	Clement Mignon					
3	1	Gustav Iden	Gold				
4	5	Joe Skipper					
5	3	Kristian Blummenfelt	Bronze				
6	7	Leon Chevalier					
7	8	Magnus Ditlev					
8	4	Max Neumann					
9	10	Patrick Lange					
10	2	Sam Laidlow	Silber				
11	6	Sebastian Kienle					
12							
13							

So funktioniert ERSTERWERT() in Excel



Die Formel ist deutlich kürzer und einfacher zu verstehen. Des Weiteren könnte ERSTERWERT() ohne Probleme mit weiteren Medaillenzuweisungen – falls es diese geben sollte – ergänzt werden.

## Wert aus Zelle übernehmen mit Excel-Funktion ÜBERNEHMEN()

Mit der Excel-Funktion ÜBERNEHMEN() schneiden Sie aus einer Liste einen gewünschten Bereich aus und stellen diesen gesondert in Ihrer Tabelle dar. In der Kombination mit anderen Funktionen ergeben sich daraus flexible Anwendungsmöglichkeiten.

Zuletzt geändert am 18.03.2026



Die Excel-Funktion **ÜBERNEHMEN()** ist ein beeindruckendes Werkzeug, mit dem Sie Daten aus einer vorhandenen Tabelle schnell und dynamisch übernehmen können – und zwar, ohne dass Sie komplizierte Formeln verwenden müssen.

### Parameter der Excel-Funktion ÜBERNEHMEN()

Die Funktion hat die folgende Syntax und Parameter:

**=ÜBERNEHMEN(array; rows; [columns])**

- **Array**: Bereich der Liste, aus der Sie Daten übernehmen wollen
- **Rows**: positive Zahl = Anzahl der Zeilen von oben; negative Zahl = Anzahl der Zeilen von unten
- **Columns** (optional): positive Zahl = Anzahl der Spalten von links; negative Zahl = Anzahl Spalten von rechts

### Beispiel für die Excel-Funktion ÜBERNEHMEN()

Das folgende Beispiel für die Funktion ÜBERNEHMEN() zeigt, was sie bewirkt. Aus der Liste sollen automatisch immer die ersten drei Zeilen an eine andere Stelle der Arbeitsmappe gefiltert und ausgegeben werden.

	A	B	C	D
1	<b>Filiale</b>	<b>Umsatz</b>	<b>Gewinn</b>	
2	Bremen	4.546.883	681.536	
3	Dortmund	6.076.390	797.547	
4	Dresden	3.162.985	-364.555	
5	Düsseldorf	9.993.777	-486.752	
6	Köln	5.079.531	290.676	
7	Leipzig	3.475.239	92.146	
8	Leverkusen	7.737.745	811.807	
9	Magdeburg	7.131.057	696.894	
10	München	9.042.466	768.108	
11	Nürnberg	4.730.670	-102.704	
12	Stuttgart	4.186.561	1.071.218	
13	Würzburg	6.804.710	600.544	
14				
15				

Beispiel: Liste für das Übernehmen von Werten

Erfassen Sie hierzu einfach die folgende Formel in der Zelle, in der Sie die ersten drei Zeilen ausgeben oder anzeigen wollen:

**=ÜBERNEHMEN(A2:C13; 3; 3)**

- Array: Bereich der Liste = **A2:C13**
- Rows: Anzahl der Zeilen von oben = **3** (die ersten 3 Zeilen der Liste)
- Columns (optional): Anzahl der Spalten von links = **3** (für alle 3 Spalten der Liste)

Die Überschriftenzeile mit Filiale, Umsatz und Gewinn wurde kopiert und in Zelle F2 die Funktion ÜBERNEHMEN() eingegeben.

E2		⌵	:	✕	✓	<i>f<sub>x</sub></i>	=ÜBERNEHMEN(A2:C13;3;3)
	A	B	C	D	E	F	G
1	Filiale	Umsatz	Gewinn		Filiale	Umsatz	Gewinn
2	Bremen	4.546.883	681.536		Bremen	4.546.883	681.536
3	Dortmund	6.076.390	797.547		Dortmund	6.076.390	797.547
4	Dresden	3.162.985	-364.555		Dresden	3.162.985	-364.555
5	Düsseldorf	9.993.777	-486.752				
6	Köln	5.079.531	290.676				
7	Leipzig	3.475.239	92.146				
8	Leverkusen	7.737.745	811.807				
9	Magdeburg	7.131.057	696.894				
10	München	9.042.466	768.108				
11	Nürnberg	4.730.670	-102.704				
12	Stuttgart	4.186.561	1.071.218				
13	Würzburg	6.804.710	600.544				

Ergebnis der Excel-Funktion ÜBERNEHMEN()

Ändern sich die Listeneinträge, dann werden die Inhalte durch die Formel =ÜBERNEHMEN() automatisch angepasst.

## Anwendungsmöglichkeiten für ÜBERNEHMEN()

Der praktische Nutzen von ÜBERNEHMEN() zeigt sich, wenn Sie Ihre Daten in der Ausgangsliste entsprechend sortieren. Mit einem einfachen Klick können Sie die Liste nach verschiedenen Kriterien sortieren und sofort aktualisierte Ergebnisse sehen.

Sie können sich beispielsweise immer die drei Filialen mit dem höchsten Gewinn anzeigen lassen, wenn Sie die Ausgangsliste absteigend nach dem Gewinn sortieren.

Die Kombination aus **SORTIERENNACH()** und **ÜBERNEHMEN()** bietet Ihnen flexible Möglichkeiten zur Datenauswertung und Datenaufbereitung, wie die folgende Abbildung zeigt.

E2 <span>✕</span> <span>✓</span> <span>fx</span> =ÜBERNEHMEN(SORTIERENNACH(A2:C13;C2:C13;-1);3;3)						
	A	B	C	D	E	F
1	Filiale	Umsatz	Gewinn		Filiale	Umsatz
2	Bremen	4.546.883	681.536		Stuttgart	4.186.561
3	Dortmund	6.076.390	797.547		Leverkusen	7.737.745
4	Dresden	3.162.985	-364.555		Dortmund	6.076.390
5	Düsseldorf	9.993.777	-486.752			
6	Köln	5.079.531	290.676			
7	Leipzig	3.475.239	92.146			
8	Leverkusen	7.737.745	811.807			
9	Magdeburg	7.131.057	696.894			
10	München	9.042.466	768.108			
11	Nürnberg	4.730.670	-102.704			
12	Stuttgart	4.186.561	1.071.218			
13	Würzburg	6.804.710	600.544			

Kombination der Excel-Funktionen SORTIERENNACH() und ÜBERNEHMEN()

Wie spielen die Funktionen zusammen? Zunächst die innere Funktion:

### **SORTIERENNACH(A2:C13; C2:C13; -1)**

Der Bereich A2:C13 wird nach der Spalte mit dem Gewinn (C2:C13) absteigend (-1) sortiert. Mehr zu dieser Funktion lesen Sie im [Excel-Tipp zu SORTIERENNACH\(\)](#).

Aus dieser sortierten Liste werden dann die ersten drei Zeilen und drei Spalten übernommen. Die Funktion dafür lautet:

### **=ÜBERNEHMEN(SORTIERENNACH(A2:C13; C2:C13; -1); 3; 3)**

Das Ergebnis ist in der vorigen Abbildung im Bereich E2:G4 dargestellt.

**Wichtig:** Bei ÜBERNEHMEN() handelt es sich um eine Array-Funktion. Sie müssen deshalb darauf achten, dass nach unten und nach rechts ausreichend leere Zellen sind. Ansonsten erhalten Sie die Fehlermeldung #ÜBERLAUF!

## Excel-Werte auflisten, die bestimmtes Kriterium erfüllen

Mit dieser Formel können Sie in Excel alle Werte größer als 1000 aus einem bestimmten Bereich auslesen und in einer Spalte auflisten. Sie funktioniert natürlich auch mit anderen Auswahlkriterien.

Zuletzt geändert am 18.03.2026



### Wofür braucht es die selektive Wertauswahl?

Sie wollen einen großen Datensatz durchforsten, um Werte zu finden, die eine bestimmte Bedingung erfüllen. Zum Beispiel wollen Sie alle Werte aus einem Zellbereich extrahieren, die größer als der Schwellenwert 1000 sind. Warum könnte das wichtig sein?

- Stellen Sie sich vor, Sie analysieren Verkaufszahlen und möchten schnell die wichtigsten Transaktionen, die über einem bestimmten Betrag liegen, identifizieren.
- Oder Sie haben es mit wissenschaftlichen Messdaten zu tun und möchten nur die besonders stark abweichenden Messwerte für weitere Auswertungen heranziehen.

Diese Art der Datenextraktion ist unerlässlich, um große Datenmengen zu reduzieren und sich auf die relevanten Informationen zu konzentrieren.

### Werte extrahieren mit Excel

In Excel gibt es verschiedene Wege, um diese Aufgabe zu lösen – von der einfachen Filterfunktion bis hin zu dynamischen Funktionen, die automatisch die gewünschten Daten ausgeben.

In diesem Excel-Tipp erfahren Sie, mit welcher Formel Sie die gewünschten Daten aus einem Bereich filtern und an einer anderen Stelle auflisten können.

In der folgenden Abbildung sehen Sie einen Bereich mit Umsatzzahlen. Sie wollen die Zahlen, die größer als 1000 sind, filtern und an einer anderen Stelle untereinander auflisten.

	A	B	C
1	904	1224	771
2	1439	1341	777
3	939	314	545
4	749	654	724
5	1430	872	666
6	672	244	50
7	1146	500	879
8	900	333	189
9	1225	234	687
10	120	456	832
11	1116	276	1145
12	750	31	923
13	691	193	1164
14	992	390	1349
15	926	521	766
16	563	346	961
17	1025	272	649

Tabelle mit unterschiedlichen Zahlenwerten

## Formel zur Ausgabe von Werten mit Auswahlkriterium

Diese Aufgabe können Sie mit der folgenden Formel lösen:

**=ZUSPALTE(WENN(A1:C17>1000; A1:C17; NV()); 2)**

Die Formel geht davon aus, dass die auszuwertenden Daten im Bereich **A1:C17** stehen.

E1						<b>=ZUSPALTE(WENN(A1:C17&gt;1000;A1:C17;NV());2)</b>
	A	B	C	D	E	
1	904	1224	771		1224	
2	1439	1341	777		1439	
3	939	314	545		1341	
4	749	654	724		1430	
5	1430	872	666		1146	
6	672	244	50		1225	
7	1146	500	879		1116	
8	900	333	189		1145	
9	1225	234	687		1164	
10	120	456	832		1349	
11	1116	276	1145		1025	
12	750	31	923			
13	691	193	1164			
14	992	390	1349			
15	926	521	766			
16	563	346	961			
17	1025	272	649			

Formel zur Ausgabe von Zahlen, die ein definierten Grenzwert überschreiten (zeilenweises Durchsuchen)

## Was leistet die Formel?

Mit der Funktion **WENN(A1:C17>1000; A1:C17; NV())** wird zunächst jede Zelle im Bereich A1:C17 daraufhin untersucht, ob der Wert größer 1000 ist oder nicht. Ist der Wert größer 1000, dann wird der entsprechende Wert, sonst der Fehlerwert **NV()** für die jeweilige Zelle ausgegeben.

Dieses Ergebnis wird dann an die Funktion **ZUSPALTE()** weitergeben:

**=ZUSPALTE(WENN(A1:C17>1000; A1:C17; NV()); 2).**



Mit der Funktion ZUSPALTE() können Sie einen Bereich in eine Spalte umwandeln. Dabei können Sie entscheiden, ob Sie den Bereich zeilen- oder spaltenweise umwandeln und ob Fehler und leere Einträge ignoriert werden sollen.

Mit dem **Argument 2** in der Funktion ZUSPALTE(WENN(A1:C17>1000; A1:C17; NV()); **2**); legen Sie fest, dass die Fehlerwerte ignoriert werden sollen. Und die Fehlerwerte entsprechen im Beispiel Zellen, deren Wert nicht größer als 1000 ist.

Mit der Formel wird der Datenbereich **zeilenweise** durchsucht und die Werte werden entsprechend aufgelistet. Wollen Sie den Datenbereich **spaltenweise** durchsuchen und auflisten, dann müssen Sie die Formel wie folgt anpassen:

=ZUSPALTE(WENN(A1:C17>1000; A1:C17; NV()); 2; **WAHR**)

	A	B	C	D	E	F
1	904	1224	771		1439	
2	1439	1341	777		1430	
3	939	314	545		1146	
4	749	654	724		1225	
5	1430	872	666		1116	
6	672	244	50		1025	
7	1146	500	879		1224	
8	900	333	189		1341	
9	1225	234	687		1145	
10	120	456	832		1164	
11	1116	276	1145		1349	
12	750	31	923			
13	691	193	1164			
14	992	390	1349			
15	926	521	766			
16	563	346	961			
17	1025	272	649			

Formel zur Ausgabe von Zahlen, die ein definierten Grenzwert überschreiten (spaltenweises Durchsuchen)

## Funktionen für die geordnete Ergebnisdarstellung

Wollen Sie die Ergebnisse in Spalte E nun auch noch sortieren, ergänzen Sie die Formel folgendermaßen:

**=SORTIEREN(ZUSPALTE(WENN(A1:C17>1000; A1:C17; NV()); 2))**

	A	B	C	D	E	F
1	904	1224	771		1025	
2	1439	1341	777		1116	
3	939	314	545		1145	
4	749	654	724		1146	
5	1430	872	666		1164	
6	672	244	50		1224	
7	1146	500	879		1225	
8	900	333	189		1341	
9	1225	234	687		1349	
10	120	456	832		1430	
11	1116	276	1145		1439	
12	750	31	923			
13	691	193	1164			
14	992	390	1349			
15	926	521	766			
16	563	346	961			
17	1025	272	649			

*Ergebnisse der Werteauswahl sortieren*

Kommen bestimmte Werte mehrfach in Ihrer Tabelle vor und wollen Sie diese aber nur einmal in der Ergebnisspalte aufführen, setzen Sie die Funktion **EINDEUTIG()** ein. Die Formel lautet dann:

**=EINDEUTIG(ZUSPALTE(WENN(A1:C17>1000; A1:C17; NV()); 2))**

E1	=EINDEUTIG(ZUSPALTE(WENN(A1:C17>1000;A1:C17;NV());2))					
	A	B	C	D	E	F
1	904	1224	771		1224	
2	1439	1341	777		1439	
3	939	314	545		1341	
4	749	654	724		1430	
5	1430	872	666		1146	
6	672	244	50		1225	
7	1146	500	1439		1116	
8	900	333	189		1145	
9	1225	234	1439		1164	
10	120	456	832		1349	
11	1116	276	1145		1025	
12	750	31	923			
13	691	193	1164			
14	992	390	1349			
15	926	521	766			
16	563	346	961			
17	1025	272	649			

Werte, die mehrfach vorkommen, nur einmal auflisten

## Zelladressen unter Auswahlbedingung in Excel anzeigen

Mit dieser Formel können Sie in Excel die Zelladressen aller Werte größer als 1000 aus einem bestimmten Bereich auslesen und auflisten. Diese bedingte Ausgabe von Zelladressen funktioniert genauso mit anderen Auswahlkriterien.

Zuletzt geändert am 18.03.2026



Sie wollen nicht nur wissen, welche Werte in einem bestimmten Bereich bestimmte Bedingungen erfüllen, sondern auch, wo diese Werte sich befinden – also welche Zelladressen sie haben?

**Beispiel:** Sie wollen alle Zellen identifizieren, deren Werte größer als 1000 sind.

Das Wissen über die genauen Zelladressen ist wichtig, wenn Sie gezielt weiter mit diesen Zellen arbeiten oder sie in anderen Formeln verwenden möchten.

In diesem Beitrag zeigen wir Ihnen, wie Sie in Excel die Zelladressen von Werten über 1000 ermitteln und auflisten können. Und das funktioniert natürlich genauso mit allen anderen gesuchten Werten.

### Beispiel für die Auswahl von Zelladressen

In der folgenden Abbildung sehen Sie einen Bereich mit Umsatzwerten. Sie sollen die Zelladressen, welche größer als 1000 sind, an einer anderen Stelle untereinander auflisten.

	A	B	C
1	904	1224	771
2	1439	1341	777
3	939	314	545
4	749	654	724
5	1430	872	666
6	672	244	50
7	1146	500	879
8	900	333	189
9	1225	234	687
10	120	456	832
11	1116	276	1145
12	750	31	923
13	691	193	1164
14	992	390	1349
15	926	521	766
16	563	346	961
17	1025	272	649

Beispiel: Liste mit Werten zur Auswertung

## Formel zur bedingten Ausgabe von Zelladressen

Sie können diese Aufgabe mit der folgenden Formel lösen:

**=ZUSPALTE(WENN(A1:C17>1000; ADRESSE(ZEILE(A1:C17); SPALTE(A1:C17)); NV()); 2)**

Die Formel geht hierbei von der Voraussetzung aus, dass A1:C17 der Bereich ist, in dem die zu prüfenden Werte stehen.

E1	=ZUSPALTE(WENN(A1:C17>1000;ADRESSE(ZEILE(A1:C17);SPALTE(A1:C17));NV());2)							
	A	B	C	D	E	F	G	H
1	904	1224	771		\$B\$1			
2	1439	1341	777		\$A\$2			
3	939	314	545		\$B\$2			
4	749	654	724		\$A\$5			
5	1430	872	666		\$A\$7			
6	672	244	50		\$A\$9			
7	1146	500	879		\$A\$11			
8	900	333	189		\$C\$11			
9	1225	234	687		\$C\$13			
10	120	456	832		\$C\$14			
11	1116	276	1145		\$A\$17			
12	750	31	923					
13	691	193	1164					
14	992	390	1349					
15	926	521	766					
16	563	346	961					
17	1025	272	649					

Ausgabe der Zelladresse unter Angabe von Bedingungen

## Die Parameter und Funktionen der Formel

Die Formel überprüft jede Zelle im Bereich A1:C17 und liefert entweder die Zelladresse oder den Fehlerwert #NV.

Das bewirken die einzelnen Komponenten:

- **A1:C17>1000**: Diese Bedingung prüft, ob die Werte im Bereich A1:C17 größer als 1000 sind. Wenn der Wert größer als 1000 ist, wird **WAHR** zurückgegeben, andernfalls **FALSCH**.
- **ZEILE(A1:C17)** liefert für jede Zelle im Bereich die entsprechende Zeilennummer (zum Beispiel 1 für Zellen in der ersten Zeile, 2 für Zellen in der zweiten usw.).
- **SPALTE(A1:C17)** gibt für jede Zelle die Spaltennummer zurück (1 für Spalte A, 2 für Spalte B usw.).

- **ADRESSE(ZEILE(A1:C17); SPALTE(A1:C17))** kombiniert die Zeilen- und Spaltennummern und gibt die Adresse der jeweiligen Zelle als Text zurück (zum Beispiel \$A\$1 für die Zelle in der ersten Zeile und ersten Spalte). Dieser Teil der Formel sorgt dafür, dass die Zelladressen aller Zellen, deren Werte größer als 1000 sind, als Text zurückgegeben werden.
- **NV()**: Wenn der Wert in einer Zelle kleiner oder gleich 1000 ist, gibt die Funktion WENN den Fehlerwert #NV (Nicht Verfügbar) zurück. Dies sorgt dafür, dass Sie später bei der Verwendung von ZUSPALTE keine leeren oder falschen Werte aufgelistet bekommen.

Zusammenfassend sorgt diese Kombination aus WENN, ADRESSE, ZEILE und SPALTE dafür, dass Sie entweder die Zelladresse für Zellen mit einem Wert größer als 1000 erhalten oder den Fehlerwert #NV, wenn der Wert die Bedingung nicht erfüllt.

Die Zelladressen und Fehlerwerte werden anschließend an die Funktion ZUSPALTE() weitergegeben:

**=ZUSPALTE(WENN(A1:C17>1000; ADRESSE(ZEILE(A1:C17); SPALTE(A1:C17))); NV()); 2)**

## Was die Funktion ZUSPALTE() bewirkt

Die Funktion **ZUSPALTE()** wandelt ein mehrdimensionales Array, wie in diesem Fall den Bereich A1:C17, in eine einspaltige Liste um. Sie sorgt dafür, dass alle Zelladressen oder Fehlerwerte aus dem ursprünglichen mehrspaltigen Bereich in einer einzelnen Spalte angezeigt werden.

Hierbei gibt der zweite Parameter in ZUSPALTE(...; **2**) an, dass die Funktion ZUSPALTE() keine Fehlerwerte berücksichtigt. Fehlerwerte entsprechen in unserem Beispiel Zellen, deren Werte kleiner als 1000 sind und daher nicht weiter berücksichtigt werden sollen.

Standardmäßig werden die Daten spaltenweise gelesen und umgewandelt. Das bedeutet, Excel wird zunächst die erste Spalte, dann die zweite und schließlich die dritte Spalte durchlaufen, um die Zelladressen oder Fehlerwerte zu extrahieren.

## Zeilenweise Auswertung der Liste mit ZUSPALTE()

Wollen Sie die Daten **zeilenweise** im Datenbereich lesen und umwandeln, dann müssen Sie die Formel wie folgt anpassen:

**=ZUSPALTE(WENN(A1:C17>1000; ADRESSE(ZEILE(A1:C17); SPALTE(A1:C17))); NV()); 2; WAHR)**

	A	B	C	D	E	F	G	H
1	904	1224	771		\$A\$2			
2	1439	1341	777		\$A\$5			
3	939	314	545		\$A\$7			
4	749	654	724		\$A\$9			
5	1430	872	666		\$A\$11			
6	672	244	50		\$A\$17			
7	1146	500	879		\$B\$1			
8	900	333	189		\$B\$2			
9	1225	234	687		\$C\$11			
10	120	456	832		\$C\$13			
11	1116	276	1145		\$C\$14			
12	750	31	923					
13	691	193	1164					
14	992	390	1349					
15	926	521	766					
16	563	346	961					
17	1025	272	649					

*Zeilenweise Auswertung des Bereichs mit der Funktion ZUSPALTE()*



# Suchfunktion für die exakte Suche in Excel nutzen

Die Suchfunktion in Excel beinhaltet einige Einstellungen, mit denen Sie steuern, wie exakt das Ergebnis mit Ihrem Suchbegriff übereinstimmen soll. Welche Einstellungen können Sie in der Suchfunktion vornehmen?

Zuletzt geändert am 18.03.2026



Excel verfügt über eine **Suchfunktion**, mit der Sie ein Arbeitsblatt, eine Arbeitsmappe oder einen bestimmten Zellbereich nach bestimmten Suchkriterien durchsuchen können.

Im Standardfall wird hierbei der erfasste Suchbegriff mit den Inhalten des Suchbereichs verglichen. Kommt der Suchbegriff in den jeweiligen Zellen vor, dann werden diese als Ergebnis angezeigt.

Dies trifft auch zu, wenn der Suchbegriff nur mit einem Teil des Zellinhaltes übereinstimmt.

## Beispiel für die Suchfunktion in Excel

Die Zellinhalte in der folgenden Abbildung sollen als Beispiel dienen.


	A	B
1	Business Wissen	
2	Business	
3	Business wissen	
4	BUSINESS WISSEN	
5	Business Wissen	
6	business	
7		
8		
9		

Beispiel: Zellinhalte mit unterschiedlichen Einträgen

Zunächst wollen Sie nach dem Wort **Business** suchen.

Öffnen Sie das Dialogfeld **Suchen und Ersetzen**, indem Sie die Tastenkombination **Strg + F** drücken.


Erfassen Sie hier unter **Suchen nach:** den Suchbegriff **Business** und klicken Sie anschließend mit der linken Maustaste auf die Schaltfläche **Alle suchen**.



Suchfunktion in Excel

Excel listet Ihnen anschließend im unteren Bereich des Dialogfelds alle gefundenen Zellen auf, in denen der Suchbegriff vorkommt.

**Tipp:** Per Klick mit der linken Maustaste auf den entsprechenden Eintrag in der Trefferliste des Suchdialogs (siehe roter Kasten in der folgenden Abbildung) aktivieren Sie automatisch die entsprechende Zelle im Tabellenblatt.




Ergebnis der Suchfunktion in Excel

Wollen Sie jetzt nur die Zellen finden, in denen der Suchbegriff exakt vorkommt, das heißt ohne weitere zusätzliche Ziffern und Zeichen, dann aktivieren Sie das Kontrollkästchen **Gesamten Zelleninhalt vergleichen** und klicken wieder mit der linken Maustaste auf die Schaltfläche **Alle suchen**.


Sie bekommen dann nur noch die Zellen A2 und A6 als Ergebnis aufgelistet, da dies die einzigen beiden Zellen sind, in denen nur der gesuchte Suchbegriff steht.

Allerdings wird bei dieser Auswahl noch keine Rücksicht auf die Groß- und Kleinschreibung genommen.



Suchergebnis mit exakter Übereinstimmung ohne Groß-/Kleinschreibung

Wollen Sie die Suche noch verfeinern und weiter einschränken, indem auch die Groß- und Kleinschreibung beachtet wird, dann aktivieren Sie zusätzlich noch das Kontrollkästchen **Groß-/Kleinschreibung beachten** und klicken mit der linken Maustaste auf die Schaltfläche **Alle suchen**.



Suchergebnis mit exakter Übereinstimmung mit Groß-/Kleinschreibung

Nutzen Sie als  
**Premium-Mitglied**  
alle  
**Handbuch-Kapitel**  
mit mehr als  
3.000 Checklisten und Excel-Vorlagen

**Jetzt anmelden**

[www.business-wissen.de/anmelden/](http://www.business-wissen.de/anmelden/)

## Impressum

b-WISE GmbH Business Wissen Information Service  
Bismarckstraße 21  
76133 Karlsruhe  
DEUTSCHLAND

[service@business-wissen.de](mailto:service@business-wissen.de)  
Telefon +49 721 18397-0

Copyright 2026, b-wise GmbH, All Rights Reserved